Contents

Chapter 1 • DATA Step Basics ................................................................. 1
  DATA Step Examples ....................................................................... 2
  The DATA Step and CAS ................................................................. 3
  Requirements for Running the DATA Step in CAS ......................... 4
  Controlling Where the DATA Step Runs ....................................... 5
  DATA Step Processing Modes ....................................................... 7
  BY-Group Processing in the CAS DATA Step ................................. 11
  SAS Language Elements Support for the CAS DATA Step .............. 14
  CAS DATA Step Action .................................................................... 19
  Data Types in the CAS DATA Step .................................................. 19
  Rules for Names in the CAS DATA Step .......................................... 19
  Automatic DATA Step Variables .................................................... 19
  Avoiding Data Truncation ............................................................... 22

Chapter 2 • DATA Step Examples ....................................................... 29
  Run the DATA Step in CAS ............................................................. 31
  Run the DATA Step in CAS on a CAS Table .................................... 32
  Load a CSV File into CAS and Save as a CAS Table ...................... 34
  Convert a CAS Table to a SAS Data Set .......................................... 35
  Combine Tables Using the SET and BY Statements ....................... 36
  Sum a Variable across an Entire Table .......................................... 39
  Score an In-Memory CAS Table ..................................................... 41
  Create a DATA Step View .............................................................. 42
  Create a VARCHAR Variable Using the LENGTH Statement .......... 45
  Create a VARCHAR Using an ARRAY Statement ............................ 47
  Impute Missing Values Using the LAG Function ............................. 48
  Index CHAR and VARCHAR Character Strings ............................... 50
  Implicit Type Conversion .............................................................. 52
  Group and Order a CAS Table Using the PARTITION= Data Set Option ... 54
  Group and Order Rows Using the DATA Step BY Statement .......... 56
  Group Data by Two Variables ....................................................... 59
  Create a DATA Step Hash Object with VARCHAR Key and Data Variables ... 60
  Run the DATA Step in a Single Thread .......................................... 63
  View DATA Step Processing Information Using Automatic Variables ... 64
  Use the Macro Facility to Generate CAS DATA Step Code ............. 66
  DATA Step Program Walk-through ............................................... 68
Chapter 1

DATA Step Basics

DATA Step Examples .................................................. 2
  Set Up Code for Examples ........................................ 2
  Common Tasks ......................................................... 2
The DATA Step and CAS .............................................. 3
  Role of the DATA Step with CAS ................................. 3
  Why Run the DATA Step in CAS? ................................. 4
Requirements for Running the DATA Step in CAS ................. 4
  Requirements ......................................................... 4
Controlling Where the DATA Step Runs ............................. 5
  Determining Where the DATA Step Is Running ................. 5
  Controlling Where the DATA Step Runs by Default .......... 6
DATA Step Processing Modes ....................................... 7
  Single versus Multithreaded Processing ........................ 7
  Distributed versus Non-Distributed Data ......................... 7
  Scenarios .................................................................. 8
  Multithreaded DATA Step Processing on Distributed Data .... 8
  Multithreaded DATA Step Processing on a Single Machine .... 9
  Single-threaded DATA Step Processing ........................... 9
  Controlling Threaded Processing .................................. 9
  Inter-row Dependencies in DATA Step Processing ............. 11
BY-Group Processing in the CAS DATA Step ....................... 11
  How CAS Groups Data with BY Variables ...................... 11
  Identifying the First and Last Rows in BY-Groups .......... 12
  How BY Variables Affect Multithreaded DATA Step Execution 12
  BY-Group Processing versus Partitioning ..................... 13
  Other Methods for Partitioning, Grouping, and Ordering .... 14
SAS Language Elements Support for the CAS DATA Step ........ 14
  Supported Language Elements in the DATA Step ............. 14
  Restrictions for Language Elements in the DATA Step ....... 15
  Macro Processing .................................................... 17
CAS DATA Step Action ................................................. 19
Data Types in the CAS DATA Step .................................. 19
Rules for Names in the CAS DATA Step ........................... 19
Automatic DATA Step Variables .................................... 19
  Overview .............................................................. 19
DATA Step Examples

Set Up Code for Examples

The following code connects you to a CAS server, starts a CAS session, creates a CAS engine libref, and associates SAS librefs with caslibs. You might not need to manually connect to the CAS server (for example, if you are running SAS Studio, which connects to CAS automatically). If you do need to connect to the server, uncomment the OPTIONS statement below and follow the numbered instructions.

```sas
/* options cashost="cloud.example.com" casport=5570 */      /* */
cas casauto sessopts=(caslib='casuser');                    /* */
libname mycas cas;                                          /* */
caslib _all_ assign;                                        /* */
```

1. Specify the host name of your CAS server in quotation marks in the CASHOST= option. Specify the port number that the server is running on in the CASPORT= option.

2. Start a CAS session named Casauto and specify Casuser as the active caslib (see “Caslibs, Files, and Tables” in SAS Cloud Analytic Services: Fundamentals for more information about the Casuser caslib).

3. If you have not already done so, create a CAS engine libref.

4. Assign SAS librefs to existing caslibs to allow caslib names to appear in the user interface.

Common Tasks

Run the DATA step in CAS

```sas
cas casauto;
libname <libref> cas;
data <libref>.<table-name>;
   put 'Hello from ' _hostname_' thread # ' _threadid_;
x=1;
run;
```
Run the DATA step on a table in CAS

```sas
cas casauto;
libname <libref> cas;
data <libref>..<table-name>;
set <libref>..<table-name>;
<statements>;
runchasasauto;
```

Create CHAR and VARCHAR variables

```sas
cas casauto;
libname <libref> cas;
data <libref>..<table-name>;
length <variable-name> varchar(n);
length <variable-name> $;
runchasasauto;
```

Add and promote a table

```sas
cas casauto;
libname <libref> cas;
data <libref>..<table-name>(copies=1 promote=yes);
set <libref>..<table-name>;
runchasasauto;
```

Merge CAS tables

```sas
cas casauto;
libname <libref> cas;
data <libref>..<table-name>;
merge <libref>..<table-1><libref>..<table-2>;
by <variable-name>;
runchasasauto;
```

---

The DATA Step and CAS

**Role of the DATA Step with CAS**

The DATA step can run in both Cloud Analytic Services (CAS) and in SAS. CAS is the cloud-based run-time environment that enables multithreaded, parallel DATA step execution.

How you run the DATA step in CAS is almost identical to how you run it in SAS with the traditional DATA step. The DATA step in CAS supports most of the same language elements and provides much of the same functionality.

The DATA step serves two roles in the CAS environment:

- **Loading a SAS data set to CAS** – processing is done in SAS in a single thread in the SAS client session. A CAS engine libref is used to load the data after it has been processed in the SAS DATA step.

  ```sas
data mycas.cars;
set sashelp.cars;
runchasasauto;
```

- **Processing in CAS** – processing is done in CAS in multiple threads. Processing is done on an in-memory CAS table. Note that this is a separate process from loading the data and can be done only in a separate DATA step on data already loaded to CAS.
Why Run the DATA Step in CAS?

Big data running in a single thread is slow. The DATA step runs faster in CAS because it can run in multiple threads, in parallel. In CAS, hundreds of threads are available, across several machines. When you program your DATA step to run in CAS, all of those threads become available for you to use to improve the performance with massively parallel processing. Parallel processing speeds the execution of big data by starting the DATA step on multiple machines and dividing processing workload among threads on these machines.

Requirements for Running the DATA Step in CAS

Requirements

You must perform the following steps to run the DATA step in CAS:

- start a CAS session
- create a CAS engine libref using the LIBNAME statement with the CAS option
- specify the same CAS engine libref on both the input and output data
- use only CAS-supported language elements
- ensure that there is at least one variable in the output table (CAS does not support zero-column tables)

Example Code 1.1 Example of DATA Step That Meets All the Requirements for Running in CAS

```sas
data mycas.cars2;
  set mycas.cars;
run;
```

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

For information about the language elements that are supported in CAS, see “SAS Language Elements Support for the CAS DATA Step” on page 14.
Controlling Where the DATA Step Runs

Determining Where the DATA Step Is Running

You can tell where your DATA step is running by looking at the SAS log output. The following note is printed to the SAS log when the DATA step runs in CAS.

```
NOTE: Running DATA step in Cloud Analytic Services.
NOTE: The DATA step will run in multiple threads.
NOTE: There were 322 observations read from the table BASEBALL in caslib CASUSERHDFS().
NOTE: The table baseball2 in caslib CASUSERHDFS() has 322 observations and 24 variables.
NOTE: DATA statement used (Total process time):
  real time           0.22 seconds
  cpu time            0.00 seconds
```

Where the DATA step runs is driven by where your tables are stored:

- If your data sets are stored in SAS, the DATA step runs in SAS.
- If your tables are stored in CAS, the DATA step runs in CAS.
- If your tables are stored in a mix of locations, then the DATA step runs in SAS.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

In the following example, the DATA step runs in SAS because the input is a SAS data set. The resulting table is stored in CAS.

**Example Code 1.2  DATA Step Running in SAS**

```
libname mycas cas;

data mycas.baseball;
  set sashelp.baseball;
run;
```

**Output 1.1  Log Output**

```
NOTE: There were 322 observations read from the data set SASHELP.BASEBALL.
NOTE: The data set MYCAS.BASEBALL has 322 observations and 24 variables.
```

Here, the DATA step runs in CAS because both tables are stored in CAS. The two-level name beginning with the CAS engine libref, Mycas, indicates that the tables are CAS tables.

**Example Code 1.3  DATA Step Running in CAS**

```
data mycas.baseball2;
  set mycas.baseball;
run;
```
NOTE: Running DATA step in Cloud Analytic Services.
NOTE: The DATA step will run in multiple threads.

When the MSGLEVEL=i option is specified in the OPTIONS statement, more detailed information is displayed in the log.

**Example Code 1.4**   **DATA Step Running in CAS with MSGLEVEL=i Set**

```sas
options msglevel=i;

data mycas.baseball2;
  set mycas.baseball(where=(Team="New York");
run;
```

**Output 1.3**   **Log Output**

NOTE: The where clause is being processed by Cloud Analytic Services.
NOTE: Running DATA step in Cloud Analytic Services.
NOTE: The DATA step will run in multiple threads.

If you use a language element that is not supported in the CAS DATA step, then the DATA step will automatically run in SAS rather than generate an error.

The following code shows how a CAS DATA step that contains an unsupported language element will run in SAS rather than display an error. The WHERE clause in the DATA statement is invalid for a CAS DATA step. In order to run in CAS, the WHERE clause can be used only in the SET statement.

**Example Code 1.5**   **DATA Step Invalid in CAS Runs in SAS**

```sas
data mycas.baseball2(where=(Team="New York");
  set mycas.baseball;
run;
```

Since MSGLEVEL=i was specified, the message in the log offers the information.

**Output 1.4**   **Log Output**

NOTE: The WHERE data set option on the DATA statement is not supported with DATA step in Cloud Analytic Services.

**Controlling Where the DATA Step Runs by Default**

You can use the **DSCAS | NODSCAS** system option to control whether the DATA step runs in CAS. The default behavior is for the DATA step to run in CAS by default. This means that the DATA step first tries to run in CAS, and if it fails, it runs in SAS.
The **DATA** step will fail to run in CAS if any of the requirements for running in CAS are not met. See “Requirements for Running the **DATA** Step in CAS” on page 4 for a list of these requirements.

If you want the **DATA** step to run in SAS only, without first trying to run in CAS, you can specify the **NODASCAS** system option:

```plaintext
options nodscas;

data mycas.test;     /* This **DATA** step runs in SAS */
set mycas.air;
run;
```

Knowing and controlling where the **DATA** step is running is important for the following reasons:

- performance can be enhanced or degraded depending on the size of the table and where you run the **DATA** step. Running the **DATA** step in multiple threads in CAS is recommended for large tables.
- programs that rely on values retained from one row to the next (calculations that have inter-row dependencies) behave differently depending on where the **DATA** step runs. See “Inter-row Dependencies in **DATA** Step Processing” on page 11.
- grouping and ordering of data is handled differently in the multithreaded CAS **DATA** step. Because each thread operates independently on a portion of the table, there is no global ordering of the table. See “BY-Group Processing in the CAS **DATA** Step” on page 11.

---

**DATA Step Processing Modes**

**Single versus Multithreaded Processing**

**Single-threaded**

the **DATA** step runs in SAS or in CAS in a single thread on an entire, non-distributed table. Running the **DATA** step in a single thread is not recommended for very large tables.

**Multithreaded (default)**

the **DATA** step runs in CAS in multiple threads in parallel. Each **DATA** step thread operates on a portion of the table. The **DATA** step can run in multiple threads in a single-machine (non-distributed) server or in a distributed server.

**Distributed versus Non-Distributed Data**

**Distributed server**

a multi-node server environment in which CAS distributes the rows of a table across multiple physical machines. **DATA** step processing is distributed across the nodes so that the processing power of multiple CPUs running in parallel is combined and runs on the partitioned table. This is ideal for very large tables that require a large amount of processing power.

**Single-Machine server**

a single-machine server in a non-distributed environment that is configured to run as the server. When you run the **DATA** step in multiple threads in a single-machine
server, the processing is performed on each CPU core on that machine. This enables multithreaded, parallel DATA step execution.

For more information about CAS server architecture, see *SAS Cloud Analytic Services: Fundamentals*.

**Scenarios**

Scenarios for DATA step processing in CAS include the following:

- “Multithreaded DATA Step Processing on Distributed Data”
- “Multithreaded DATA Step Processing on a Single Machine”
- “Single-threaded DATA Step Processing”

**Multithreaded DATA Step Processing on Distributed Data**

A multithreaded DATA step is one in which the processing workload is shared among multiple, identical DATA steps that run concurrently on different parts of the table. In a distributed data environment, the table schema is duplicated on every node and each node holds different rows of the table.

In a distributed CAS server environment, DATA step processing is distributed across worker nodes so that the processing power of multiple CPUs running in parallel is combined. When the DATA step runs in a distributed server, it runs in multiple threads on each node. On each node, CAS divides the table into groups or blocks of data, and the DATA step runs on each group, allocating one DATA step thread per group. This is ideal for very large tables that require a large amount of processing power.

The DATA step runs by default in multiple threads on every available node.

*Note:* Because the rows of a table are distributed across multiple machines and partitioned on each node in this way, each DATA step thread processes only a portion of the table. This separation of table rows might affect how some functions that have inter-row dependencies behave. See “Inter-row Dependencies in DATA Step Processing” on page 11 for more information.
**Multithreaded DATA Step Processing on a Single Machine**

In non-distributed CAS environments, a single-machine is configured to run as the CAS server. The single-machine contains multiple CPU cores to support multithreaded, parallel processing. When the DATA step runs in multiple threads on a single-machine server, the entire table is loaded onto this single machine. CAS partitions the table based on the number of cores available and each DATA step thread runs on each partition. Splitting the table among threads in this way speeds up processing. Instead of having one DATA step thread work on the entire table, there are many DATA steps working on the table in parallel.

**Single-threaded DATA Step Processing**

Single-threaded DATA step processing occurs when the DATA step runs in a single thread on an entire, non-distributed, non-partitioned table. You can run the DATA step in a single thread, even when there are multiple CAS servers available. When you run the DATA step in a single thread in a distributed environment, processing is localized to a single machine regardless of the number of available worker nodes.

To specify that the DATA step runs on a single thread, specify the SINGLE= option in the DATA statement. For an example, see “Run the DATA Step in a Single Thread” on page 63.

Running the DATA step in a single thread is not recommended for very large tables. Single-threaded processing might be useful when you need to compute sums across multiple rows. See “Sum a Variable across an Entire Table” on page 39 for an example that shows how single-threaded DATA step processing might be needed.

**CAUTION:**
If a single-threaded DATA step processes large tables, not only does the program run slowly, the machine executing the single thread can run out of disk or memory resources.

**Controlling Threaded Processing**

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2. After setting up for examples, you can run the examples in this section.

**Default Processing**

A DATA step that runs in CAS, runs in all available threads by default. If the DATA step has no input table, then it runs in a single thread.

**Example Code 1.6**  The DATA Step Runs in Multiple Threads by Default When There Is an Input Table

```plaintext
data mycas.test;
  set mycas.test;
  put 'The DATA step is running in '_nthreads_ 'threads';
run;
```
Output 1.5  Log Output

NOTE: Running DATA step in Cloud Analytic Services.
NOTE: The DATA step will run in multiple threads.
The DATA step is running in 192 threads

Example Code 1.7  A DATA Step with No Input CAS Table Runs in a Single Thread

data mycas.test;
   put 'The DATA step is running in ' _nthreads_ 'thread';
run;

NOTE: Running DATA step in Cloud Analytic Services.
NOTE: The DATA step has no input data set and will run in a single thread.
The DATA step is running in 1 thread

The number of available threads depends on the number of active server nodes and CPU cores available on each node. The DATA step typically runs one DATA step thread per core.

For example, if your CAS configuration consists of four worker nodes and each node supports 32 threads (or cores), then the DATA step will run in 128 threads. The number of DATA step threads equals the number of nodes times the number of threads running on each node.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

Controlling the Number of Threads
You can control whether the DATA step runs in a single thread or in all available threads by specifying the SINGLE= option in the DATA statement.

• SINGLE=YES specifies that the DATA step runs in a single thread.

    data mycas.class / single=yes;
    set mycas.class;
    run;

• SINGLE=NO specifies that the DATA step runs in all available threads (this is the default behavior):

    data mycas.class / single=no;
    set mycas.class;
    run;

• SINGLE=NOINPUT causes a DATA step without input tables to run in a single thread instead of in all available threads. This is the default value for a DATA step with no input tables when you do not specify the SESSREF= option.

• THREADS= specifies the number of threads that the DATA step runs in on each node. A value of zero means that the DATA step will run in the maximum number of threads allowed. This option is for fine tuning and it might show different results depending on the number of CPU cores on each node.
Inter-row Dependencies in DATA Step Processing

A multithreaded DATA step can reduce run time and improve performance. But functions or statements that rely on information from one row to another might return unexpected results when they are used in a multithreaded DATA step.

For example, the RETAIN statement has inter-row dependencies because it retains a value from one row to the next. But, in a multithreaded DATA step, the rows in a table are not necessarily processed together in the same DATA step thread, nor is their order guaranteed.

Therefore, the RETAIN statement can no longer operate on the entire table, which is needed for this type of operation.

Here are some SAS language elements that have inter-row dependent operations:

- RETAIN statement
- DIF and LAG functions
- .FIRST and .LAST automatic DATA step variables (see “How SAS Identifies the Beginning and End of a BY Group” in SAS DATA Step Statements: Reference)
- “Sum Statement” in SAS DATA Step Statements: Reference
- Temporary arrays

The calculations performed by these types of statements are often referred to as full-table operations. Full-table operations require reading the entire contents of a table as a whole, rather than just reading selected portions.

Full-table operations can be performed in two steps: the first step involves performing a local operation, such as computing a partial sum. The second step involves performing a global operation, such as computing the final sum from a series of partial sums.

This two-step operation works well because it uses the processing power of all the threads to process the entire table and a single thread to process the smaller table subset.

For an example of this type of program, see Sum a Variable Across an Entire Table. on page 39

For more information about temporary arrays, see ARRAY Statement.

BY-Group Processing in the CAS DATA Step

How CAS Groups Data with BY Variables

When the DATA step runs in a distributed server, CAS distributes the input table across multiple servers. Each server gets only a portion of the rows in the table.

The BY statement in a SAS DATA step divides table data into groups of rows that share the same values of the BY variables. When you use the BY statement in a distributed server, CAS groups the rows based on the first BY variable. When CAS distributes the table across multiple nodes, it keeps the BY groups intact. That is, rows that share the same BY variables are stored together on a single machine.

Here are some key points to remember about BY-group processing in the CAS DATA step:
CAS groups and distributes table rows based on the values of the first BY variable, then orders the rows within each BY group based on all the BY variables.

On the CAS server, there is no guarantee of global ordering between BY groups. Each DATA step thread can group and order only the rows that are in the same data partition.

If a format is applied to the first BY variable, then CAS creates BY groups based on the formatted values of the first BY variable (rather than the raw values). If there is not a format applied to the first BY variable, CAS creates BY groups based on the raw values.

The DATA step orders data within each BY group based on the raw values of all the BY variables. When a BY group is defined, the DATA step orders only the rows within that BY group.

The DATA step runs in multiple threads on each node, one thread per BY group.

Formats can have a noticeable impact on performance when they are used with BY variables. This is because CAS groups and distributes rows based on the format of the first BY variable. If a high cardinality format is applied to the first BY variable, then more threads will be used. Conversely, if a low cardinality format is applied to the first BY variable, then fewer threads will be used.

See the example in “How BY Variables Affect Multithreaded DATA Step Execution” on page 12 for an illustration of low and high-cardinality BY-groups.

For information about BY-Group processing in SAS, see “BY-Group Processing in the DATA Step” in SAS Language Reference: Concepts.

Identifying the First and Last Rows in BY-Groups

In the DATA step, SAS identifies the beginning and end of each BY group by creating two temporary variables. One temporary variable is created for the first row of the BY group and another temporary variable is created for the last row of the BY group. CAS creates a FIRST. and LAST. variable for every node. These variables can be used to manipulate how data is grouped, subdivided, and ordered.

Here are two examples that use the .FIRST variable to group rows in a table:

• “Group and Order Rows Using the DATA Step BY Statement” on page 56
• “Group Data by Two Variables” on page 59

For information about how the DATA step identifies BY groups and how you can manipulate them, see FIRST and LAST DATA Step Variables.

How BY Variables Affect Multithreaded DATA Step Execution

Because an entire BY group is operated on by a single DATA step thread, fewer threads are used if there are fewer BY groups. Fewer BY groups are created when the first BY variable has a low cardinality or is represented by a format with a low cardinality. Therefore, choose your first BY variable wisely and be aware of formats and their cardinality. You might need to assign or reassign formats to your BY variables to enhance performance.

Here are some examples of low and high cardinality formats:
Formats with a High Cardinality
- DATEw.
- DATETIMEw.d
- DDMMYYw.

Formats with a Low Cardinality
- QTRw.
- MONNAME.
- WEEKDAYw.

The following image shows the relationship between the cardinality of the first BY variable and the number of threads that are spawned by the DATA step in BY-group processing.

Note: This image depicts a Bygroup column that shows the start of each BY group. For simplicity, the full code used to create this column is not shown in the image. To see the full example, see “View DATA Step Processing Information Using Automatic Variables” on page 64.

Note: The DESCENDING option in the BY statement is not supported in a DATA step that is running in CAS.

**BY-Group Processing versus Partitioning**

BY-group processing in the CAS DATA step is an ad hoc grouping of data that is useful for performing one-time DATA step computations on small to medium tables. When you perform BY-group processing on a CAS table, the DATA step creates a temporary copy of the in-memory table that exists for the duration of the DATA step job. Once the DATA step finishes processing, the temporary table and all its BY groups disappear. Future data access to BY groups requires that the table is re-grouped and re-distributed each time. In
other words, BY-group processing in the CAS DATA step does not affect the structure or distribution of the original in-memory table. BY-group processing is, therefore, more suitable for instantaneous DATA step computation.

The partitioning feature in CAS provides a more permanent and efficient solution for grouping and managing data. The work required for partitioning does not have to be repeated as it does when using BY groups in the DATA step. When you partition a table as you load it using the Partition action, it becomes a partitioned in-memory table that can be accessed by subsequent operations.

To partition CAS tables, you can use the partition action. See “Partition a Table Using the Partition Action” in *SAS Viya: System Programming Guide* for examples and syntax for using the Partition action in CAS.

**Other Methods for Partitioning, Grouping, and Ordering**

Here are some other methods for ordering and grouping data:

- **PARTITION action**
- **ORDERBY=** and **GROUPBY=** options in the PROC CASUTIL LOAD statement
- **PARTITION=** and **ORDERBY=** data set options in the SAS DATA statement

For an example of using the BY statement as described in this section, see “Group and Order Rows Using the DATA Step BY Statement” on page 56.

---

**SAS Language Elements Support for the CAS DATA Step**

**Supported Language Elements in the DATA Step**

Not all SAS language elements are supported in CAS. Below are links to the reference documentation for the SAS DATA Step language. Each document contains a “By Category” table that lists the CAS supported language elements under the CAS table entry:

- “Data Set Options by Category” in *SAS Data Set Options: Reference*
- “Formats by Category” in *SAS Formats and Informats: Reference* and “Informats by Category” in *SAS Formats and Informats: Reference.*
- “SAS Functions and CALL Routines by Category” in *SAS Functions and CALL Routines: Reference*
- “DATA Step Statements by Category” in *SAS DATA Step Statements: Reference*
Support for CAS is also listed on the individual language element syntax pages for all the DATA step language elements. The support is indicated in the categories field as shown in the example document below:

**Figure 1.2** Example Documentation Showing CAS Support on Individual Syntax Page


**Restrictions for Language Elements in the DATA Step**

The following table contains restrictions on the use of some DATA step language elements in CAS.
<table>
<thead>
<tr>
<th>Feature</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>APPEND= data set option</td>
<td>You cannot set both the PROMOTE= and the APPEND= data set options to YES at the same time (in the same DATA step). You can set either PROMOTE= or APPEND= to YES. An error results if you set both to YES at the same time. For more information, see “APPEND= Data Set Option” in SAS Cloud Analytic Services: User’s Guide.</td>
</tr>
<tr>
<td>BY variables</td>
<td>BY variables are supported in the DATA step in CAS but there are some differences between how SAS groups variables and how CAS groups them. See “BY-Group Processing in the CAS DATA Step” on page 11 for more information.</td>
</tr>
<tr>
<td>BY statement, DESCENDING option</td>
<td>The DESCENDING option in the BY statement is not supported in a DATA step that is running in CAS.</td>
</tr>
<tr>
<td>FORMAT procedure</td>
<td>User-defined formats created with PROC FORMAT must exist both in SAS and in CAS when using them in a DATA step that is running in CAS. If the format is not available to SAS, the DATA step will fail before it tries to run in CAS. For more information, see “FORMAT Procedure” in Base SAS Procedures Guide.</td>
</tr>
<tr>
<td>Formats</td>
<td>These formats are not supported in a DATA step that is running in CAS:</td>
</tr>
<tr>
<td></td>
<td>- WORDS</td>
</tr>
<tr>
<td></td>
<td>- WORDSF</td>
</tr>
<tr>
<td></td>
<td>- WORDDATE (use NLDATES instead)</td>
</tr>
<tr>
<td></td>
<td>- WEEKDATE (use NLDATEW instead)</td>
</tr>
<tr>
<td></td>
<td>The NLDATEW format and the NLDATE format automatically translate the date values to the language specified in your LOCALE= setting.</td>
</tr>
<tr>
<td>Functions</td>
<td>Not all functions are supported in a DATA step that is running in CAS. See Functions and CALL Routines by Category for a list of supported functions.</td>
</tr>
<tr>
<td>Functions and Session Encoding</td>
<td>The default encoding method for sessions in CAS is UTF-8. In a UTF-8 environment, data can contain multi-byte characters that are equivalent to more than 1 byte. Some functions that parse character strings to calculate length and positional values might return unexpected results. For example, the LENGTH function returns the number of bytes rather than the number of characters. See “About Session Encoding” on page 22 for information about session encoding affects behavior. For an example showing how some functions behave, see “Index CHAR and VARCHAR Character Strings” on page 50 for an example of this behavior.</td>
</tr>
<tr>
<td>INFILE, INPUT, and DATALINES statements</td>
<td>These language elements are not supported in a DATA step that is processing in CAS. You can use these statements in a SAS client session and then upload the results to CAS.</td>
</tr>
</tbody>
</table>
### Macros

The SYMGET and SYMPUT functions are not supported in a DATA step that is running in CAS. For more information about macro processing in SAS, see “Macro Processing” on page 17.

### PROMOTE= data set option

You can set either PROMOTE= or APPEND= to YES. An error results if you set both to YES at the same time.

### MODIFY, REMOVE, and REPLACE statements

These statements are not supported in a DATA step that is running in CAS, and you cannot use them to modify an in-memory CAS table. You can use these statements in a SAS client session and then upload the results to CAS.

### VARCHAR data type

There are some restrictions on using the VARCHAR data type. See “CAS LIBNAME Statement” in SAS Cloud Analytic Services: User’s Guide for more information.

### WHERE= data set option

CAS does not support the WHERE= data set option for output tables. If you specify the WHERE= data set option on the output data set, the DATA step automatically runs in SAS rather than in CAS. Specify the MSGLEVEL=i option to see more processing information in the log.

### zero-column tables

CAS does not support zero-column tables. This is different from the DATA step in SAS, which does support zero-column tables. You will get an error if you try to create a table in CAS that has no variables.

---

### Macro Processing

#### Support for Macro Processing

Macro code itself does not execute on the server. It is precompiled and runs in the SAS client session. The code that the macro generates can be made to run in CAS, however. See “Macro Examples” on page 18.

The following DATA Step Functions for Macros are not supported in a DATA step that is running in CAS:

- SYMGET
- SYMGETN

The following DATA Step CALL Routines for Macros are not supported in a DATA step that is running in CAS:

- CALL SYMPUT Routine
- CALL SYMPUTN Routine
- CALL SYMPUTX Routine

These restrictions in macro processing do not apply to the SAS client. These functions are supported in a DATA step that is processing in SAS:

**Example Code 1.8  SYMGET Is Supported in SAS Client Session Processing**

```sas
data test;
  x = SYMGET('SYSDATE9');
  put x;
run;
```
But, they are not supported in a DATA step that is running in CAS:

**Example Code 1.9  SYMGET Is Not Supported in a DATA Step That Is Processing in CAS**

```sas
data mycas.test;
  x = SYMGET('SYSDATE9');
  put x;
run;
```

**Output 1.6  Log Output Showing Error When SYMGET Is Used in a CAS DATA Step**

```
ERROR: The function SYMGET is unknown, or cannot be accessed.
```

For more information about macro processing in SAS, see “Using the Macro Facility in SAS Viya” in *SAS Macro Language: Reference*.

**Macro Examples**

You can use SAS macros in a variety of ways with the CAS server environment. For example, you can create macros that generate SAS code that executes in CAS. But, the macro code itself does not execute on the server.

Here are some examples showing how to use macros to execute the CAS procedure, the CASUTIL procedure, and the CAS DATA step.

**Example Code 1.10  Create a Macro for Uploading a Sashelp Data Set into CAS**

```sas
%macro loadit(name=);
  proc casutil;
    load data=sashelp.&name
    outcaslib='casuser'
    replace;
  run; quit;
%mend loadit;

%loadit(name=cars)
```

**Example Code 1.11  Create a Macro for Saving a Sashelp Data Set in CAS**

```sas
%macro loadsaveit(name=);
  proc casutil;
    load data=sashelp.&name
    outcaslib='casuser'
    replace;
    save casdata="&name" replace;
  run; quit;
%mend loadsaveit;

%loadsaveit(name=cars)
```

**Example Code 1.12  Create a Macro for Dropping a Table from Memory**

```sas
%macro dropit(name=);
  proc casutil incaslib="casuser";
    droptable casdata="&name"
    list files;
  run; quit;
%mend dropit;
```
Example Code 1.13  Create a Macro for Getting Server Information Using the serverStatus Action

```ssa
%macro aboutServer;
    serverStatus;
%mend;
proc cas;
    %aboutServer;
run;
```

**CAS DATA Step Action**

CAS actions are requests that are sent by the user to the server to perform a task. You can use CAS actions to submit a DATA step. CAS actions are submitted within a PROC CAS block.

For more information about the DATA step actions available in CAS, see “DATA Step Action Set” in *SAS Viya: System Programming Guide*.

**Data Types in the CAS DATA Step**

For information about data types, see “Data Types” in *SAS Cloud Analytic Services: User’s Guide*.

**Rules for Names in the CAS DATA Step**

For information about rules for names in CAS, see “Variable Names and Data Set Names in CAS Engine” in *SAS Cloud Analytic Services: User’s Guide*.

**Automatic DATA Step Variables**

**Overview**

Automatic variables are system variables that are created automatically by the SAS DATA step and saved in memory. They are not included in the output data set being created. The values of automatic variables are retained from one iteration of the DATA step to the next, rather than set to missing. For an example that uses automatic variables, see “View DATA Step Processing Information Using Automatic Variables” on page 64.

The following automatic variables are created by every DATA step that runs in a SAS client session or in CAS. For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.
_ERROR_

is valid in the SAS client session and in the CAS server session.

_ERROR_ is initialized to 0. It is set to 1 when an error is encountered. Errors include input data errors, conversion errors, or math errors, such as division by 0. You can use the value of this variable to help locate errors in data records and to print error messages to the SAS log.

In the following example, the IF statement writes to the SAS log during each iteration of the DATA step. The DATA step writes the contents of the input record when an error is encountered:

```
data mycas.test;
  input x y;
  if _error_ = 1 then
    put "****** Error in row " _n_ " ******";
  datalines;
  1 1
  2 3
  3 g
  4 4
;```

NOTE: Invalid data for y in line 64 3-3.
****** Error in row 3  ******
RULE:      ----+----1----+----2----+----3----+----4----+----5--
64         3 g
x=3 y=. _ERROR_=1 _N_=3
NOTE: The data set MYCAS.TEST has 4 observations and 2 variables.

_N_

is an internal system variable that counts the iterations of the DATA step as it automatically loops through the rows of an input data set. The _N_ variable is initially set to 1 and increments by 1 each time the DATA step loops past the DATA statement. The DATA step loops past the DATA statement for every row that it encounters in the input data. Because the DATA step is a built-in loop that iterates through each row in a table, the _N_ variable can be used as a counter variable.

Note: The DATA step creates and stores the _N_ automatic variable internally but it does not include the variable in the output data set.

You can use the PUT statement with the _ALL_ argument to print the values of _N_ and _ERROR_ to the SAS log.

In a multi-node server session, _N_ returns the number of DATA step iterations (table rows) per worker node, per thread. For example, if an eight-row table is distributed across four worker nodes (with each node holding 2 rows), then the log displays all rows processed on each node:

```
data mycas.test;
  set mycas.test;
  put "This is iteration (row) # " _n_ " on " _hostname_; 
run;
```
returns the name of the worker node or host that the DATA step is running on. In a DATA step that is processing in SAS, _HOSTNAME_ is the name of the machine SAS is running on.

```sas
data mycas.bigcars(where=(weight>5000));
  set sashelp.cars;
  keep make weight;
run;

data mycas.bigcars;
  set mycas.bigcars;
  keep make weight;
  put _hostname_;
run;
```

returns the number of DATA step threads running in a server session. In a DATA step that is processing in SAS, _NTHREADS_ is always 1.

```sas
data mycas.bigcars(where=(weight>5800));
  set sashelp.cars;
  keep make weight;
run;
```

returns the name of the worker node or host that the DATA step is running on. In a DATA step that is processing in SAS, _HOSTNAME_ is the name of the machine SAS is running on.
returns the number that is associated with the thread that the DATA step is running in a server session. In a DATA step that is processing in SAS, _THREADID_ is always 1.

data mycas.bigcars(where=(weight>5800));
set sashelp.cars;
keep make weight;
run;
data mycas.bigcars;
set mycas.bigcars;
  put "The ThreadID is " _threadid_;
run;

Avoiding Data Truncation

About Session Encoding

If your SAS session uses an encoding method other than UTF-8, SAS automatically transcodes data that you move from SAS to CAS, and vice versa. Transcoding is the automatic process by which SAS converts data from one session encoding to another.

The CAS server uses UTF-8 session encoding. Your local SAS session can use any number of different encoding methods depending on your locale and the settings that are configured for your site. The default SAS session encoding is WLatin1 for Windows and Latin1 for UNIX.

The number of bytes for a character in your local SAS session encoding can be different from the number of bytes for the same character in CAS. For example, the characters é and ë in a Latin1 session encoding require only one byte for storage but in UTF-8 encoding these same characters require 2 bytes for storage. This is because UTF-8 is a multi-byte character set (MBCS) while Latin1 is a single-byte character set (SBCS). Therefore, transcoding can result in data truncation.

Complete information about session encoding and how you can manage transcoding issues in SAS can be found in here:

Specific information about the VARCHAR data type can be found in “VARCHAR Data Type” in SAS Cloud Analytic Services: User’s Guide.

**Determine the Session Encoding**

To determine session encoding for your SAS session, specify the PROC OPTIONS statement:

```sas
proc options option=encoding; run;
```

The session encoding information is printed in the SAS log.

To determine session encoding for a specific SAS data set, specify the CONTENTS procedure:

```sas
proc contents data=sashelp.manage; run;
```

**Examples of Truncation Errors**

**Overview**

When you create a variable in a CAS DATA step, the information about the length of the variable is calculated in the SAS session then sent to CAS for processing. CAS creates a table with columns based on the information from SAS and then reads the values into the table using UTF-8 character encoding. When this happens, truncation of the data can occur if your local session is not compatible with the UTF-8 session. Or, you might get a transcoding error in the SAS log.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

**Example 1**

In the following example, the local SAS session encoding is EUC-CN (Chinese) and the DATA step runs in CAS. The code is first partially compiled in the SAS client where information about the length of the variables is determined and then sent to CAS for processing. CAS creates the table based on these column widths and then reads in the values for the variable using UTF-8 character encoding.

In this example, because the EUC-CN character set requires 2 bytes to store characters, the length of the CHAR is set to 8 bytes (to enable saving 4 EUC-CN characters). SAS sends this length information to CAS and CAS creates a table column with a width of 8 bytes. CAS then reads the values into the column using UTF-8 character encoding. The values require 3 bytes per character in UTF-8. The column width is defined with a length of only 8 bytes, so the CHAR data is truncated.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
data mycas.example;
    length vc varchar(4);
    c = ‘我的例子’;
run;
```

**Solution:** Explicitly define the length for the CHAR to ensure that it is long enough to accommodate the length when the characters are converted to UTF-8 encoding. If your local session encoding uses a DBCS, then set the CHAR length to be equal to the
original length times 2. In the solution below, the length of the CHAR is set to 12 bytes, enough to hold 4 multibyte characters.

```sas
data mycas.example;
  length vc varchar(4);
  length c $8; /* in UTF-8 this variables might require up to 4 bytes */
  vc = '我的例子';
  c = '我的例子';
run;
```

**Example 2**
In the following example, the DATA step is running in CAS. The CHAR variable contains an extended Latin1 character. The extended Latin1 character, ä, requires 1 byte in Latin1 and 2 bytes in UTF-8. SAS compiles the DATA step in the local SAS session and sends information about the variable lengths to CAS. CAS reads in the variable values using UTF-8. The output data is truncated.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
data mycas.example;
  length c $4;
  length vc varchar(*);
  c = 'äbcd'; /* length in Latin1 is 4 bytes */
  vc = 'äbcd'; /* length in CAS UTF-8 is 5 */
run;
```

**Solution:** Increase the length of the CHAR variable.

```sas
data mycas.example2;
  length c $5;
  length vc varchar(*);
  c = 'äbcd'; /* length in Latin1 is 4 bytes */
  vc = 'äbcd'; /* length in CAS UTF-8 is 5 */
run;
```

```sas
proc print data=mycas.example2;
```

<table>
<thead>
<tr>
<th>Obs</th>
<th>c</th>
<th>vc</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>äbc</td>
<td>äbcd</td>
</tr>
</tbody>
</table>

Another solution is to change the variable type to VARCHAR:

```sas
data mycas.example2;
  length c varchar(*);
  length vc varchar(*);
  c = 'äbcd'; /* length in Latin1 is 4 bytes */
  vc = 'äbcd'; /* length in CAS UTF-8 is 5 */
run;
```

```sas
proc print data=mycas.example2;
```

<table>
<thead>
<tr>
<th>Obs</th>
<th>c</th>
<th>vc</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>äbcd</td>
<td>åbcd</td>
</tr>
</tbody>
</table>

**Note:** The NCHARMULTIPLIER=, TRANSCODE_FAIL, and ENCODING= data set options are not supported in a DATA step that runs in CAS. However, they can be used effectively in a DATA step that is loading to CAS to manage transcoding errors.
Example 3
When you save a CAS table that contains a VARCHAR variable as a SAS data set, SAS automatically converts the variable to a CHAR. You will see the following in the log.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

NOTE: One or more variables were converted because the data type is not supported by the V9 engine.

If you load a SAS data set to CAS, you might get a transcoding error due to incompatible session encodings between SAS and CAS. When you transfer SAS data sets to CAS, SAS converts the data in your local SAS session into UTF-8 data for the CAS session.

ERROR: Some character data was truncated during transcoding in the dataset MYCAS.CITIES. Use of the NCHARMULTIPLIER option is recommended.

For example, the following program might cause an error if the local SAS session encoding is Latin1:

```sas
data cities;
  city = "Sévérac-le-Château"; /* 15 Latin1 characters totaling 15 bytes */
run; /* 15 UTF-8 characters totaling 18 bytes */
data mycas.cities;
  set cities;
run;
```

To fix the transcoding error, the NCHARMULTIPLIER= data set option is specified on the output data set. This option increases the length of the transcoded variables.

```sas
data mycas.cities(NCHARMULTIPLIER=2);
  set cities;
run;
```

VARCHAR Length Conversion

Support

VARCHAR variables are not supported by the SAS V9 engine, so SAS automatically converts VARCHAR variables to CHAR variables.

If you try to store a data containing a VARCHAR in anything other than a CAS table, SAS automatically converts the VARCHAR variables to CHAR variables. When converting a variable from a VARCHAR(n) to a CHAR, SAS sets the length of the converted VARCHAR based on your local SAS session encoding. To calculate the length, SAS multiplies the current length of the VARCHAR, n, by the maximum value that a character’s length can be in the local SAS session encoding.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

Example 1
In this example, assume that the local SAS session encoding is Latin1.

```sas
proc options option=encoding; /* determine the encoding for your local session */
```
run;
data mycas.example;
  length c $10;
  length vc varchar(4);
  c = 'abcd';
  vc = 'abcd';
run;
proc contents data=mycas.example; run;

Here is the Log output for the OPTIONS procedure:

ENCODING=WLATIN1 Specifies the default character-set encoding for the SAS session.
NOTE: PROCEDURE O

Here is the PROC CONTENTS output for the original CAS table:

<table>
<thead>
<tr>
<th>#</th>
<th>Variable</th>
<th>Type</th>
<th>Len</th>
<th>Max Bytes Used</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>c</td>
<td>Char</td>
<td>10</td>
<td></td>
</tr>
<tr>
<td>2</td>
<td>vc</td>
<td>Varchar</td>
<td>4</td>
<td>4</td>
</tr>
</tbody>
</table>

data example;
  set mycas.example;
run;
proc contents data=example; run;

Here is the PROC CONTENTS output for the SAS data set with the VARCHAR variable converted to a CHAR variable. Notice how the length stays the same when the local session encoding is Latin1. SAS calculates the length of the converted VARCHAR variable by multiplying the original length by the maximum number of bytes required to store characters in the character set. Latin1 is a single-byte character set, so the length is 4 x 1, or 4 bytes.

<table>
<thead>
<tr>
<th>#</th>
<th>Variable</th>
<th>Type</th>
<th>Len</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>c</td>
<td>Char</td>
<td>10</td>
</tr>
<tr>
<td>2</td>
<td>vc</td>
<td>Char</td>
<td>4</td>
</tr>
</tbody>
</table>

**Example 2**

For this example, use the program in “Example 1” on page 25 and assume that the local session encoding is UTF-8. When the VARCHAR variable is converted to a CHAR variable, the length of the CHAR variable is 4 times the length of the largest possible character in UTF–8. The largest possible number of bytes required to store a single character in UTF-8 is 4. So, the length is 4 x 4, or 16 bytes.

Here is the output when above is run in a local UTF-8 SAS session:
More Information about SAS Session Encoding

- SAS® and UTF-8: Ultimately the Finest
- Multilingual Computing with SAS® 9.4
- Usage Note 51586: How to change the default SAS® session encoding to UTF-8
# Chapter 2

**DATA Step Examples**

<table>
<thead>
<tr>
<th>Section</th>
<th>Page</th>
</tr>
</thead>
<tbody>
<tr>
<td>Run the DATA Step in CAS</td>
<td>31</td>
</tr>
<tr>
<td>Example</td>
<td>31</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>32</td>
</tr>
<tr>
<td>See Also</td>
<td>32</td>
</tr>
<tr>
<td>Run the DATA Step in CAS on a CAS Table</td>
<td>32</td>
</tr>
<tr>
<td>Example Code</td>
<td>32</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>33</td>
</tr>
<tr>
<td>See Also</td>
<td>33</td>
</tr>
<tr>
<td>Load a CSV File into CAS and Save as a CAS Table</td>
<td>34</td>
</tr>
<tr>
<td>Example Code</td>
<td>34</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>35</td>
</tr>
<tr>
<td>See Also</td>
<td>35</td>
</tr>
<tr>
<td>Convert a CAS Table to a SAS Data Set</td>
<td>35</td>
</tr>
<tr>
<td>Example Code</td>
<td>35</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>36</td>
</tr>
<tr>
<td>See Also</td>
<td>36</td>
</tr>
<tr>
<td>Combine Tables Using the SET and BY Statements</td>
<td>36</td>
</tr>
<tr>
<td>Example Code</td>
<td>36</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>36</td>
</tr>
<tr>
<td>See Also</td>
<td>36</td>
</tr>
<tr>
<td>Sum a Variable across an Entire Table</td>
<td>39</td>
</tr>
<tr>
<td>Example Code</td>
<td>39</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>40</td>
</tr>
<tr>
<td>See Also</td>
<td>41</td>
</tr>
<tr>
<td>Score an In-Memory CAS Table</td>
<td>41</td>
</tr>
<tr>
<td>Example Code</td>
<td>41</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>42</td>
</tr>
<tr>
<td>See Also</td>
<td>42</td>
</tr>
<tr>
<td>Create a DATA Step View</td>
<td>42</td>
</tr>
<tr>
<td>Example Code</td>
<td>42</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>45</td>
</tr>
<tr>
<td>See Also</td>
<td>45</td>
</tr>
<tr>
<td>Create a VARCHAR Variable Using the LENGTH Statement</td>
<td>45</td>
</tr>
<tr>
<td>Example Code</td>
<td>45</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>47</td>
</tr>
<tr>
<td>See Also</td>
<td>47</td>
</tr>
<tr>
<td>Section</td>
<td>Page</td>
</tr>
<tr>
<td>------------------------------------------------------------------------</td>
<td>------</td>
</tr>
<tr>
<td>Create a VARCHAR Using an ARRAY Statement</td>
<td>47</td>
</tr>
<tr>
<td>Example Code</td>
<td>47</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>48</td>
</tr>
<tr>
<td>See Also</td>
<td>48</td>
</tr>
<tr>
<td>Impute Missing Values Using the LAG Function</td>
<td>48</td>
</tr>
<tr>
<td>Example Code</td>
<td>48</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>50</td>
</tr>
<tr>
<td>See Also</td>
<td>50</td>
</tr>
<tr>
<td>Index CHAR and VARCHAR Character Strings</td>
<td>50</td>
</tr>
<tr>
<td>Example Code</td>
<td>50</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>52</td>
</tr>
<tr>
<td>See Also</td>
<td>52</td>
</tr>
<tr>
<td>Implicit Type Conversion</td>
<td>52</td>
</tr>
<tr>
<td>Example Code</td>
<td>52</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>54</td>
</tr>
<tr>
<td>See Also</td>
<td>54</td>
</tr>
<tr>
<td>Group and Order a CAS Table Using the PARTITION= Data Set Option</td>
<td>54</td>
</tr>
<tr>
<td>Example Code</td>
<td>54</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>56</td>
</tr>
<tr>
<td>See Also</td>
<td>56</td>
</tr>
<tr>
<td>Group and Order Rows Using the DATA Step BY Statement</td>
<td>56</td>
</tr>
<tr>
<td>Example Code</td>
<td>56</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>58</td>
</tr>
<tr>
<td>See Also</td>
<td>58</td>
</tr>
<tr>
<td>Group Data by Two Variables</td>
<td>59</td>
</tr>
<tr>
<td>Example Code</td>
<td>59</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>60</td>
</tr>
<tr>
<td>See Also</td>
<td>60</td>
</tr>
<tr>
<td>Create a DATA Step Hash Object with VARCHAR Key and Data Variables</td>
<td>60</td>
</tr>
<tr>
<td>Example Code</td>
<td>60</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>62</td>
</tr>
<tr>
<td>See Also</td>
<td>62</td>
</tr>
<tr>
<td>Run the DATA Step in a Single Thread</td>
<td>63</td>
</tr>
<tr>
<td>Example Code</td>
<td>63</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>63</td>
</tr>
<tr>
<td>See Also</td>
<td>63</td>
</tr>
<tr>
<td>View DATA Step Processing Information Using Automatic Variables</td>
<td>64</td>
</tr>
<tr>
<td>Example Code</td>
<td>64</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>66</td>
</tr>
<tr>
<td>See Also</td>
<td>66</td>
</tr>
<tr>
<td>Use the Macro Facility to Generate CAS DATA Step Code</td>
<td>66</td>
</tr>
<tr>
<td>Example Code</td>
<td>66</td>
</tr>
<tr>
<td>See Also</td>
<td>67</td>
</tr>
<tr>
<td>DATA Step Program Walk-through</td>
<td>68</td>
</tr>
<tr>
<td>Example Code</td>
<td>68</td>
</tr>
<tr>
<td>Key Ideas</td>
<td>73</td>
</tr>
<tr>
<td>See Also</td>
<td>73</td>
</tr>
</tbody>
</table>
Run the DATA Step in CAS

Example

The following example shows how to run the DATA step in CAS. For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sql
data mycas.hello;                       /*1*/
  put 'Hello from ' _hostname_         /*2*/
    'thread #'   _threadid_;         /*3*/
    x=1;                                 /*4*/
run;
```

1  A CAS engine libref is specified in the DATA statement to create an in-memory CAS table named mycas.hello.

2  The _HOSTNAME_ automatic DATA step variable is specified in a PUT Statement to demonstrate how the DATA step runs in CAS in a distributed environment.

3  The _THREADID_ automatic DATA step variable is specified to demonstrate multithreaded processing in CAS.

4  CAS tables require at least one variable. An assignment statement is used to create a variable.

Output 2.1  Log Output Showing Multithreaded DATA Step Processing in a Distributed CAS Environment

NOTE: Running DATA step in Cloud Analytic Services.
Hello from server001 thread #2
Hello from server001 thread #3
Hello from server001 thread #4
Hello from server001 thread #1
NOTE: The table hello in caslib CASUSER(username) has 4 observations and 1 variables...
Key Ideas

- Running a DATA step in CAS means that the DATA step code and all the statements within it are executing in CAS in a CAS server session. For information about controlling DATA processing, see “Controlling Where the DATA Step Runs” on page 5.

- For a DATA step to run in CAS, you must ensure that the requirements for running the DATA step in CAS are met. See Requirements for Running the DATA Step in CAS on page 4 for more information.

- You can also use the CASUTIL procedure to load your SAS data sets into CAS. This method is usually more efficient for transferring large data sets to CAS. For more information, see “CASUTIL Procedure” in SAS Cloud Analytic Services: User’s Guide for more information.

- Not all language elements are supported in a CAS DATA step. See “SAS Language Elements Support for the CAS DATA Step” on page 14 for information about what language elements are supported in the CAS DATA step.

See Also

- “Requirements for Running the DATA Step in CAS”.
- “Automatic DATA Step Variables” on page 19 (_THREADID_ and _HOSTNAME_)
- “_ALL_” in SAS Cloud Analytic Services: User’s Guide
- “SESSOPTS=(session-option(s))” in SAS Cloud Analytic Services: User’s Guide

Run the DATA Step in CAS on a CAS Table

Example Code

To run the DATA step in CAS using input data, both the input and output must be CAS tables and both tables must use the same CAS engine libref. In the following example, the CASUTIL Procedure is used to load a SAS data set into CAS as an in-memory table. The SAS data set name is preceded by the CAS engine libref, mycas.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
proc casutil outcaslib='casuser';
   load data=sashelp.cars;                /* 1 */
run;

data mycas.cars2 / sessref=casauto;       /* 2 */
   set mycas.cars;                        /* 3 */
```
if mpg_city > 25 then eff='Y';
else eff='N';
put 'Thread number: ' _threadid_
   'on worker node ' _hostname_;
run;

1 Use PROC CASUTIL to load the Sashelp.Cars data set to CAS. This creates an in-memory CAS table named Cars. The OUTCASLIB= option loads the CAS table to the CASUSER caslib.

2 Specify a CAS table as the output data using the CAS engine libref.

3 Specify a CAS table as the input data using the CAS engine libref.

Key Ideas

• Running a DATA step in CAS means that the DATA step code and all the statements within it are executing on the CAS server. For information about how the DATA step runs in CAS, see “The DATA Step and CAS” on page 3 and “DATA Step Processing Modes” on page 7.

• For a DATA step to run in CAS, you must make sure you have met the requirements for processing in CAS on page 4.

• You can use the CASUTIL procedure to load Excel, .csv, .txt, and other types of data to CAS. See IMPORTOPTIONS= for more information. For an example of loading a .csv file to CAS, see “Load a CSV File into CAS” in SAS Cloud Analytic Services: User's Guide.

• The DATA step automatically runs in CAS when you use a CAS engine libref on the input and output data sets. It also automatically runs in CAS when you do not specify an input data set and specify output data using a CAS engine libref. For more information, see “Controlling Where the DATA Step Runs” on page 5.

• Not all language elements are supported in a CAS DATA step. See “Restrictions for Language Elements in the DATA Step” on page 15 for information about what language elements are supported in the CAS DATA step.

See Also

• “CAS Statement” in SAS Cloud Analytic Services: User’s Guide
• “CASUTIL Procedure” in SAS Cloud Analytic Services: User’s Guide
• CASLIB “_ALL_” in SAS Cloud Analytic Services: User’s Guide statement
• “SESSOPTS=(session-option(s))” in SAS Cloud Analytic Services: User’s Guide
• “The Active Caslib” in SAS Cloud Analytic Services: User’s Guide
• “Accessing Data” in SAS Cloud Analytic Services: User’s Guide
Load a CSV File into CAS and Save as a CAS Table

Example Code

The following example shows how to load a comma-separated value file (.csv) into CAS using the DATA step. This example uses the file names.csv file, which can be found here: http://support.sas.com/documentation/onlinedoc/viya/examples.htm.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
filename names url
"http://support.sas.com/documentation/onlinedoc/viya/exampledatasets/names.csv";

data mycas.names;
    infile names dsd truncover firstobs=2;
run;

proc casutil incaslib='casuser';
    save casdata='names' outcaslib='casuser' replace;
    list;
run;
```

Below is a partial view of the CSV file and its contents:

<table>
<thead>
<tr>
<th>BRTH_YR</th>
<th>GNDR</th>
<th>ETHCTY</th>
<th>NM</th>
<th>CNT</th>
<th>RNK</th>
</tr>
</thead>
<tbody>
<tr>
<td>2011</td>
<td>FEMALE</td>
<td>HISPANIC</td>
<td>GERALDINE</td>
<td>13</td>
<td>75</td>
</tr>
<tr>
<td>2011</td>
<td>FEMALE</td>
<td>HISPANIC</td>
<td>GIA</td>
<td>21</td>
<td>67</td>
</tr>
<tr>
<td>2011</td>
<td>FEMALE</td>
<td>HISPANIC</td>
<td>GIANNA</td>
<td>49</td>
<td>42</td>
</tr>
<tr>
<td>2011</td>
<td>FEMALE</td>
<td>HISPANIC</td>
<td>GISELLE</td>
<td>38</td>
<td>51</td>
</tr>
<tr>
<td>2014</td>
<td>MALE</td>
<td>WHITE NON HISPANIC</td>
<td>Youssef</td>
<td>24</td>
<td>88</td>
</tr>
<tr>
<td>2014</td>
<td>MALE</td>
<td>WHITE NON HISPANIC</td>
<td>Yusuf</td>
<td>16</td>
<td>96</td>
</tr>
<tr>
<td>2014</td>
<td>MALE</td>
<td>WHITE NON HISPANIC</td>
<td>Zachary</td>
<td>90</td>
<td>39</td>
</tr>
<tr>
<td>2014</td>
<td>MALE</td>
<td>WHITE NON HISPANIC</td>
<td>Zev</td>
<td>49</td>
<td>65</td>
</tr>
</tbody>
</table>

1 In SAS, load the external comma-separated file using the INFILE statement. Specify a CAS engine libref on the output table. The TRUNCOVER option allows SAS to correctly read in variable-length records. Variables without any values assigned are set to missing.

2 Specify the INPUT statement to list the column names and read them in as informats.

3 Save a permanent copy of the in-memory CAS table.
**Key Ideas**

- You can load external files to CAS by using an INFILE statement in the DATA step and a CAS engine libref on the output table.
- You can also load external files to CAS using the CASUTIL procedure with the LOAD FILE statement. The following examples show how to load files into CAS using PROC CASUTIL: Load a Client-Side File and Load a Server-Side File.

**See Also**

- “FILENAME Statement, URL Access Method” in *SAS Global Statements: Reference*
- “CAS Statement” in *SAS Cloud Analytic Services: User’s Guide*
- “CASUTIL Procedure” in *SAS Cloud Analytic Services: User’s Guide*
- “SESSOPTS=(session-option(s))” in *SAS Cloud Analytic Services: User’s Guide*
- “INFILE Statement” in *SAS DATA Step Statements: Reference*
- “About Informats” in *SAS Formats and Informats: Reference*
- “CASLIB Statement” in *SAS Cloud Analytic Services: User’s Guide*

---

**Convert a CAS Table to a SAS Data Set**

**Example Code**

The following example shows how to use the DATA step to convert an in-memory CAS table to a SAS data set. The first DATA step creates an in-memory CAS table that can be used for the conversion in this example. The second DATA step reads in the CAS table and creates a SAS data set in the local SAS Work library.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
   data mycas.earnings;
      Amount=1000;          /* 1 */
      Rate=.075/12;
      do month=1 to 12;
         Earned +(amount+earned)*rate;
      end;
   run;
   proc print data=mycas.earnings;
   run;

   libname mysas "pathname/";       /* 2 */
   data mysas.earnings;            /* 3 */
      set mycas.earnings;
   run;
```

1. Create variables for the table and calculate interest earned using a DO statement.
Create a libref named Mysas for storing the data set. The libref Mysas represents the physical location in which the Earnings data set is stored.

Read in the CAS table and write it out it as a SAS data set to the SAS library.

<table>
<thead>
<tr>
<th>Obs</th>
<th>Amount</th>
<th>Rate</th>
<th>month</th>
<th>Earned</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1000</td>
<td>0.0625</td>
<td>13</td>
<td>77.6326</td>
</tr>
</tbody>
</table>

**Key Ideas**

- A CAS DATA step that has an output table but no input table runs in a single thread by default. You can force the DATA step to run in multiple threads by specifying the SINGLE=NO option in the DATA statement. See “SINGLE=NO | YES | NOINPUT” in *SAS DATA Step Statements: Reference* for information about specifying the SINGLE= option.
- For more information about where the DATA step runs and controlling DATA step processing modes, see “DATA Step Processing Modes” on page 7 and “Controlling Where the DATA Step Runs” on page 5.
- The CAS DATA step works only on CAS tables.
- Not all language elements are supported in a CAS DATA step. See “Restrictions for Language Elements in the DATA Step” on page 15 for information about what language elements are supported in the CAS DATA step.

**See Also**

- “CAS Statement” in *SAS Cloud Analytic Services: User’s Guide*
- **CASLIB _ALL_** statement
- “LIBNAME Statement” in *SAS Global Statements: Reference*
- “Accessing Data” in *SAS Cloud Analytic Services: User’s Guide*
- “DATA Statement” in *SAS DATA Step Statements: Reference*

---

**Combine Tables Using the SET and BY Statements**

**Example Code**

The following example creates two data sets and then loads them into CAS tables using PROC CASUTIL. The first DATA step combines the two data sets by concatenating one to the other. This is similar to appending tables to one another. The second DATA step interleaves the two data sets by adding the BY statement. The BY variables are Common and Number.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
data animal1; /* 1 */
input Common $ Animal $ 3-8
   Number 10-11;
```
datalines;
a Ant    1
b Bird   2
c Cat    3
d Dog    4
e Eagle  5
f Frog   6
g Goose  7
h Hawk   8
i Impala 9
;
data plant1;                                 /* 2 */
input Common $ Plant $ 3-10
          Number 12-13;
datalines;
a Grape    1
c Hazelnut 2
e Indigo   3
g Jicama   4
i Kale     5
;
data append;                                 /* 3 */
   set animal1 plant1;
run;
proc print data=append;
title '1. Plant1 Appended to Animal1 in a ';
title2 'DATA Step Running in SAS'; run;
data interleaveByCommon;                     /* 4 */
   set animal1 plant1;
   by common;
run;
proc print data=interleaveByCommon;
   title '2. Animals and Plants Interleaved
          by Variable COMMON';
   title2 'in a DATA Step Running in SAS';
   run;
data interleaveByCommonNumber;               /* 5 */
   set animal1 plant1;
   by common Number;
run;
proc print data=interleaveByCommonNumber;
   title '3. Animals and Plants Interleaved
          by Common and Number';
   title2 'in a DATA Step Running in SAS';
   run;
proc casutil incaslib="casuser"              /* 6 */
   outcaslib="casuser";
   load data=animal1 replace;
   load data=plant1 replace;
run;
Create the first data set.

Create the second data set.

Append the Plant data set to the Animal data set by specifying the SET statement. No BY statement is used.

Interleave the two data sets using the BY statement with 1 variable.

Interleave the two data sets using the BY statement with 2 variables.

Load the data sets into CAS.

Combine the two data sets by specifying only the SET statement in a DATA step running in CAS. No BY statement is used.

Combine the two data sets using the BY statement with 1 variable in a CAS DATA step.

Combine the two data sets using the BY statement with 2 variables in a CAS DATA step.
**Key Ideas**

- Concatenating data sets is the combining of two or more data sets, one after the other, into a single data set. Concatenating uses the SET statement with multiple input tables. All observations from the first data set are read, followed by all observations from the second data set.

- Interleaving uses a SET statement and a BY statement to combine multiple data sets into one new data set.

- Before you can interleave data sets, the data must be sorted by the same variable or variables that are used with the BY statement that accompanies your SET statement.

- The order of rows within a BY group is not stable within CAS. Unless you explicitly specify additional BY variables to order the rows, the ordering can change from one DATA step job to the next. If you need a stable ordering, explicitly specify additional BY variables. See “How CAS Groups Data with BY Variables” on page 11.

- The BY statement combines the rows in the two data sets that have the same values for the BY variable.

- Instead of concatenating tables, you can append them and produce the same results as concatenating. To avoid reading all the records in both tables, you can append one table to another by using the APPEND= data set option.

**See Also**

- “CAS Statement” in *SAS Cloud Analytic Services: User’s Guide*
- CASLIB “_ALL_” in *SAS Cloud Analytic Services: User’s Guide* statement
- “BY Statement” in *SAS DATA Step Statements: Reference*
- “MERGE Statement” in *SAS DATA Step Statements: Reference*
- “APPEND= Data Set Option” in *SAS Cloud Analytic Services: User’s Guide*
- “Combining SAS Data Sets: Methods” in *SAS Language Reference: Concepts*

**Sum a Variable across an Entire Table**

**Example Code**

This example sums a variable across all the rows of a very large, distributed table in CAS. To do this, the example uses a 2-step process: In the first step, a DATA step runs in parallel on the “big data.” In the second step, the DATA step runs in a single thread on the smaller data set to gather the final sum. The data that is collected in the final sum is the total number of homeowners in the data set mycas.purchase.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
/* Multiple Partial sums are computed by multiple threads */
data mycas.sums;
   retain homeowners_sum;  
   /* 1 */
```
keep homeowners_sum;                   /* 2 */
set mycas.purchase end = done;         /* 3 */
if demog_ho = 1 then
  homeowners_sum + 1;                 /* 4 */
if done then output;                   /* 5 */
run;
title 'DATA Step in CAS';
proc print data=mycas.sums; run;

if demog_ho = 1 then
  homeowners_sum_all + homeowners_sum; /* 10 */
if done then output;
run;
title 'Final Sum';
proc print data=mycas.sums_all; run; /* 11 */

1 Retain the value of the homeowners_sum variable from one row to the next.
2 Keep only the homeowners_sum variable in the output table.
3 Read in the input table, mycas.purchase. Specify END= to create a flag for the conditional IF statement.
4 Create a sum of the variable Homeowners that includes the values in the rows that are being processed in that DATA step thread (on that node). Each thread calculates a partial sum for the rows that it processes. Store each sum for each thread total in homeowners_sum. See “Sum Statement” in SAS DATA Step Statements: Reference for more information.
5 When the last row is reached, SAS writes out the results.
6 Use the SINGLE=YES option to run the DATA step in a single thread.
7 Retain the value of the homeowners_sum_all variable from one row to the next.
8 Keep only the homeowners_sum_all variable in the output table.
9 Read in the input table, mycas.sums. Specify END= to create a flag for the conditional IF statement.
10 Create a total sum of Homeowners by adding the partial sums from the first DATA step. Because this DATA step runs in a single thread, the total sum of all Homeowners is calculated.
11 Print the results showing the total sum.

Key Ideas

- There is no sharing between DATA step threads.
- Each thread has its own set of rows from the table and computes values based on those rows.
- This solution speeds processing by running as much as possible in multiple threads.
- When you use a retained variable to hold a sum in a multithreaded DATA step, the variable contains only the sum of the values that that thread can see.
Score an In-Memory CAS Table

Example Code

The following example loads a SASHELP data set into CAS and then scores the data using a conditional IF statement.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
proc casutil;       /* 1 */
   load data=sashelp.baseball replace;
run;

data mycas.baseball / sessref=casauto;  /* 2 */
   set mycas.baseball(where=(team='Boston'));
   if nHits > 125
      then score=1;  /* 3 */
   keep Name Team nHits score;
run;

proc print data=mycas.baseball; run;  /* 4 */
```

1 Load the Sashelp.Baseball data set into CAS as an in-memory CAS table.
2 Run the DATA step in CAS on the table. Use the CAS engine libref (Mycas) that you created with the LIBNAME statement in Step 1 to name the input and output tables.
3 Execute the score code.
4 Run the PRINT procedure to view the output table showing the scored values.
Key Ideas

- When you run the DATA step in the CAS server, you are running it in multiple threads on in-memory data, which enables faster processing.
- To run the DATA step in a CAS server, you must specify a CAS engine libref on both the input and output data sets. For a list of requirements for running the DATA step in CAS, see “Requirements for Running the DATA Step in CAS” on page 4.

See Also

- “WHERE Statement” in SAS DATA Step Statements: Reference

Create a DATA Step View

Example Code

The following example shows how to create a DATA step view. First, an in-memory CAS table named Mycas.Class is created from the Class data set in the Sashelp library. The SET statement selects for all rows where Sex=M. A VARCHAR variable, Note, is created so that any Males who are 12 years old and taller than 60 inches are highlighted. See “Key Ideas” on page 45 for more information about DATA step views in CAS.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```
data mycas.class;                       /* 1 */
   set sashelp.class(where=(sex='M'));  /* 2 */
   varchar note;                        /* 3 */
data mycas.class;                       /* 4 */
   set sashelp.class(where=(sex='M'));  /* 5 */
   varchar note;                        /* 6 */
```
Create an in-memory CAS table named Mycas.Class.

Use the Class data set in the Sashelp library as the input data. The SET statement selects all rows where Sex=M using the WHERE statement. The rows are selected from the input data set before the output table is created in CAS.

Create a VARCHAR variable with a length of 10 characters.

Use a subsetting IF statement to select rows where Age =12 and Height is greater than 60.

If Age=12 and Height is greater than 60, then set the Note variable to “stature.”

Create a SAS DATA step view named BMI from the Mycas.Class table. The VIEW= option in the DATA statement defines the view.

Read the Mycas.Class table using the SET statement.

Calculate the BMI based on the Weight and Height variable values and store the result in the variable BMI.

Format the BMI value using the FORMAT statement.

Print the DATA step view using the PRINT procedure.

Here is the log and PROC PRINT output.

```
NOTE: DATA STEP view saved on file WORK.BMI.
NOTE: A stored DATA STEP view cannot run under a different operating system.
NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.01 seconds

    Obs  Name   Sex Age Height Weight  Note BMI
     1   Alfred  M  14  69.0   112.5   16.61
     2  Robert  M  12  64.8   128.0  21.43     stature
     3    Henry  M  14  63.5   102.5    17.87
     4   Ronald  M  15  67.0   133.0   20.83
     5    James  M  12  57.3    83.0    17.77
     6   Thomas  M  11  57.5    85.0    18.07
     7  Jeffrey  M  13  62.5    84.0    15.12
     8   William  M  15  66.5   112.0   17.80
     9    John  M  12  59.0    99.5    20.09
    10    Philip  M  16  72.0    150.0   20.34
```
To see how the SAS DATA step view handles the VARCHAR variable, the CONTENTS procedure is specified:

```sas
proc contents data=BMI;
run;
```
Key Ideas

- You can use the DATA step to create a SAS view from a CAS table. When using a CAS table as input, you must run the DATA step locally, in a Base SAS session and not in a CAS server session.
- Views can be stored only in a Base SAS library (a library created using the V9 engine). The VIEW option in the DATA statement is not supported with the DATA step in Cloud Analytic Services. This means that you cannot use a DATA step that is running in CAS to create a SAS view. However, you can create a SAS view using a DATA step that is running in SAS and that is using a CAS table as input.
- Views containing VARCHAR variables can be stored in a Base SAS library. The SAS DATA step can read and return VARCHAR variable values that are stored in SAS views. However, the DATA step cannot read and return VARCHAR variables that are contained in CAS tables or other data sources.
- A DATA step view can include VARCHAR variables but a SAS data set cannot.
- Even though a view is stored in a Base engine library, views can read data sets stored in non-Base engine libraries. You can write a view that is stored in a Base engine library and accesses data sets in a CAS engine library.
- Update views are not supported.

See Also

- “SAS Views” in SAS Language Reference: Concepts

Create a VARCHAR Variable Using the LENGTH Statement

Example Code

In the following example, the first LENGTH statement creates a VARCHAR variable, x, with a length of 30 characters. The second LENGTH statement creates a CHAR variable, y, with a length of 30 bytes.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

data mycas.string;
    length x varchar(30);
    length y $30;
    x = 'abc'; y = 'def';
run;

proc contents data=mycas.string;
run;
If you read the table into a SAS data set and then run PROC CONTENTS on the output data set, you can see how SAS converts the variable.

```sas
data string;
  set mycas.string;
run;
proc contents data=string; run ;
```

When SAS converts a VARCHAR to a CHAR in automatic data type conversion, SAS determines the length of the CHAR variable based on the SAS session encoding. If the local session encoding uses a single-byte character set, then SAS multiplies the original length by 1 to get the length of the converted CHAR variable. The length is multiplied by 1 in the calculation because 1 is the largest possible value for any character in a single-byte character set.

**Output 2.2 PROC CONTENTS Output When the Local SAS Session Encoding Is Single Byte (Latin1)**

<table>
<thead>
<tr>
<th>#</th>
<th>Variable</th>
<th>Type</th>
<th>Len</th>
<th>Max Bytes Used</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>x</td>
<td>Varch</td>
<td>30</td>
<td>30</td>
</tr>
<tr>
<td>2</td>
<td>y</td>
<td>Char</td>
<td>30</td>
<td></td>
</tr>
</tbody>
</table>

If the local session encoding is a multi-byte encoding such as UTF-8, then SAS multiplies the length times the longest possible value of any character in the character set. The example here shows the length of the converted VARCHAR when the local session encoding is UTF-8. Since 4 bytes is the largest possible value for any character in UTF-8, the length is calculated as the original length, 30, time 4 (120).

**Output 2.3 PROC CONTENTS Output When the Local SAS Session Encoding Is Multi-byte (UTF-8)**

<table>
<thead>
<tr>
<th>#</th>
<th>Variable</th>
<th>Type</th>
<th>Len</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>x</td>
<td>Char</td>
<td>120</td>
</tr>
<tr>
<td>2</td>
<td>y</td>
<td>Char</td>
<td>30</td>
</tr>
</tbody>
</table>
Key Ideas

- The VARCHAR data type is not supported by the SAS V9 engine. Therefore, you must use a CAS engine libref on the output table when creating a VARCHAR.

- To use the CAS engine with the DATA step, specify the CAS libref on the output data set. The example, “Convert a CAS Table to a SAS Data Set” on page 35, shows how to use the CAS engine to create a CAS table as output.

- The length of VARCHAR variables is determined based on the number of characters the string contains. The length of CHAR variables is determined based on the number of bytes the characters in the string requires.

- It is usually better to declare character strings as VARCHAR types than to declare them as CHAR types to avoid truncation of data in columns. Truncation can occur, for example, when multi-byte characters are stored in a CHAR data type. For example, if the DATA step converts character variables from UTF-8 to WLatin1, then the variable length (in bytes) might not be long enough and values are truncated.

- If the data is consistently short, such as in an ID column or in two-letter state abbreviations, consider using a VARCHAR with a fixed length of 16 bytes.

- With smaller variables that consist of integers with up to three digits, such as in three-letter airport codes, VARCHAR variables can increase memory use.

See Also

- “Avoiding Data Truncation” on page 22
- “VARCHAR Data Type” in SAS Cloud Analytic Services: User’s Guide
- “Create a VARCHAR Variable Using the LENGTH Statement” on page 45.
- CASLIB “_ALL_” in SAS Cloud Analytic Services: User’s Guide

Create a VARCHAR Using an ARRAY Statement

Example Code

In the following example, three VARCHAR variables are created with a maximum length for the array, Test.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

data mycas.test;
   array test[*] varchar(*) a1 a2 a3 ('a','b','c');
   put test[1]; put test[2]; put test[3];
run;
Impute Missing Values Using the LAG Function

Example Code

The following example uses the LAG function to impute missing values in data. The missing price is imputed based on a 10% increase from the price of the previous year. If the price of the previous year is also missing, then the missing price is imputed based on a 20% increase from the price of the previous year.

The following data set Mycas.Example is first created from raw input data using the DATALINES statement and written out to an in-memory CAS table. The LAG function compares values for PRICE between the rows in the Mycas.Example table. Therefore, in a distributed CAS server, where rows are located on separate machines, the LAG function does not perform as expected. To use the LAG function in a distributed CAS server, specify the SINGLE=YES statement to ensure that the data is confined to a single node.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
data mycas.example;
input county $ year price;
datalines;
1001 2001 200000
1001 2002 .
1001 2003 .
1001 2004 280000
1001 2005 310000
```
1002 2002 240000
1002 2003 270000
1002 2004 300000
1002 2005 340000
1003 2001 280000
1003 2002 300000
1003 2003 330000
1003 2004 370000
1003 2005 .
;

data mycas.example2 (drop=lag_price lag2_price lag_county lag2_county)
  / single=yes;
set mycas.example;
by county year;
  lag_price = lag(price);
  lag2_price = lag2(price);
  lag_county = lag(county);
  lag2_county = lag2(county);
  if price NE . then price_impute = price;
  else if price = . and county = lag_county and lag_price ne .
    then price_impute = lag_price*1.1;
    else if price = . and lag_price = . and county = lag2_county
      and lag2_price NE . then price_impute = lag2_price*1.2;
run;
proc print data=mycas.example2;
run;

<table>
<thead>
<tr>
<th>Obs</th>
<th>county</th>
<th>year</th>
<th>price</th>
<th>price_impute</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>1001</td>
<td>2001</td>
<td>200000</td>
<td>200000</td>
</tr>
<tr>
<td>2</td>
<td>1001</td>
<td>2002</td>
<td>.</td>
<td>220000</td>
</tr>
<tr>
<td>3</td>
<td>1001</td>
<td>2003</td>
<td>.</td>
<td>240000</td>
</tr>
<tr>
<td>4</td>
<td>1001</td>
<td>2004</td>
<td>280000</td>
<td>280000</td>
</tr>
<tr>
<td>5</td>
<td>1001</td>
<td>2005</td>
<td>310000</td>
<td>310000</td>
</tr>
<tr>
<td>6</td>
<td>1002</td>
<td>2001</td>
<td>.</td>
<td>.</td>
</tr>
<tr>
<td>7</td>
<td>1002</td>
<td>2002</td>
<td>240000</td>
<td>240000</td>
</tr>
<tr>
<td>8</td>
<td>1002</td>
<td>2003</td>
<td>270000</td>
<td>270000</td>
</tr>
<tr>
<td>9</td>
<td>1002</td>
<td>2004</td>
<td>300000</td>
<td>300000</td>
</tr>
<tr>
<td>10</td>
<td>1002</td>
<td>2005</td>
<td>340000</td>
<td>340000</td>
</tr>
<tr>
<td>11</td>
<td>1003</td>
<td>2001</td>
<td>280000</td>
<td>280000</td>
</tr>
<tr>
<td>12</td>
<td>1003</td>
<td>2002</td>
<td>300000</td>
<td>300000</td>
</tr>
<tr>
<td>13</td>
<td>1003</td>
<td>2003</td>
<td>330000</td>
<td>330000</td>
</tr>
<tr>
<td>14</td>
<td>1003</td>
<td>2004</td>
<td>370000</td>
<td>370000</td>
</tr>
<tr>
<td>15</td>
<td>1003</td>
<td>2005</td>
<td>.</td>
<td>407000</td>
</tr>
</tbody>
</table>
Key Ideas

- For more information about multithreaded processing and inter-row dependencies, see “Inter-row Dependencies in DATA Step Processing”.
- For information about the LAG function, see “LAG Function” in SAS Functions and CALL Routines: Reference.

See Also

- CASLIB _ALL_ statement
- “LAG Function” in SAS Functions and CALL Routines: Reference
- “Inter-row Dependencies in DATA Step Processing”

Index CHAR and VARCHAR Character Strings

Example Code

In the following example, the LENGTH statement creates a VARCHAR variable and a CHAR variable. Both variables are defined with a length of 10. The variables contain identical values ('abc'). The INDEX function returns the position in the string that the target value first appears. As you might expect, the INDEX function returns a value of 2 for both data types. This is because the variables x and y do not contain multi-byte characters.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

Note: Assume that the local SAS session encoding is UTF-8.

```
data mycas.string / sessref=casauto;
  length x varchar(10);
  length y $10;
  x = 'abc'; y = 'abc';
  xi = index(x,'b');
  yi = index(y,'b');
  put "xi = " xi;
  put "yi = " yi;
run;
```

Output 2.5 Log Output

```
xi = 2
yi = 2
```

However, in the next DATA step, the INDEX function is used with multi-byte Chinese characters in which each of the 3 characters takes 3 bytes. The statement `xi =`
index(x, '样') tells SAS to return the position of the character '样' in the 3-character string '榜样榜'. Even though the VARCHAR and CHAR variables contain identical string values, ('榜样榜'), the INDEX function returns different values for each type. This is because the VARCHAR data type measures data in terms of characters and the CHAR data type measures data in terms of bytes.

```sas
data mycas.string2 / sessref=casauto;
    length x varchar(10);
    length y $10;
    x = '榜样榜';  y = '榜样榜';
    xi = index(x, '样');
    yi = index(y, '样');
    put "xi = " xi;
    put "yi = " yi;
run;
```

Output 2.6 Log Output

```
xi = 2
yi = 4
```

It is important to remember that the INDEX function returns the character position when it indexes VARCHAR data types:

```
<table>
<thead>
<tr>
<th>VARCHAR string</th>
<th>‘榜样榜’</th>
</tr>
</thead>
<tbody>
<tr>
<td>character position:</td>
<td>1  2  3 (characters)</td>
</tr>
</tbody>
</table>
```

And it returns the byte position when it indexes CHAR data types.

```
<table>
<thead>
<tr>
<th>CHAR string:</th>
<th>‘榜样榜’</th>
</tr>
</thead>
<tbody>
<tr>
<td>byte position:</td>
<td>1-3 4-6 7-9 (bytes)</td>
</tr>
</tbody>
</table>
```
Key Ideas

• The INDEX function uses character-based positional indexing on VARCHAR variables, and it uses byte-based indexing on CHAR variables. The INDEX function is useful for finding a substring in a parent string. For example, to find the position of the string “hat” in “abhatcd” you specify h = index(‘abhatcd’, ‘hat’);. The function returns h = 3, because the first letter of the “hat” string is the third from the left starting with 1.

• The VARCHAR data type is not supported by the SAS V9 engine. You must use the CAS engine when creating a VARCHAR in SAS Viya. When you use the CAS engine with the DATA step, specify a CAS libref on the output data set.

• The VARCHAR data type uses character semantics. This means that the width of a column in a table is specified in character units rather than in bytes. Some characters, like the ones in this example, require more than one byte. VARCHAR variables are also varying length variables. This means that they use as many bytes as necessary to store up to 10 characters, even if some of those characters require no more than 3 or 4 bytes.

• The length of a CHAR when it is converted from a VARCHAR is based on your local session encoding. For more information, see “Avoiding Data Truncation” on page 22.

• VARCHAR variables can increase memory use when the column contains only 2- or 3-digit integer values, such as a three-letter airport. For more information about when to use a VARCHAR data type, see “When to Use a VARCHAR Data Type” in SAS Cloud Analytic Services: User’s Guide.

See Also

• “CAS Statement” in SAS Cloud Analytic Services: User’s Guide
• “_ALL_” in SAS Cloud Analytic Services: User’s Guide
• “CAS LIBNAME Statement” in SAS Cloud Analytic Services: User’s Guide
• “INDEX Function” in SAS Functions and CALL Routines: Reference
• “Data Types” in SAS Cloud Analytic Services: User’s Guide

Implicit Type Conversion

Example Code

The following example shows how a VARCHAR variable is automatically converted to a CHAR type variable when it is saved as a SAS data set. The DATA step supports the processing of VARCHAR data, but only the CAS engine supports the VARCHAR data type. The first DATA step is shown for the purposes of creating data containing a VARCHAR for this example.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
data mycas.test;                                       /*  */
  length x varchar(*);                                 */  */
```
length y varchar(20);
x="hello";
y="goodbye";
run;
proc contents data=mycas.test;
run;

data simple;  /* 2 */  
   set mycas.test;
run;

proc contents data=simple;  /* 3 */  
run;

1 Create a CAS table containing a VARCHAR for the example. The in-memory table, Mycas.Test, contains 2 VARCHAR variables.

2 Save the CAS table as a SAS data set. Note there is no CAS engine libref on the output data set.

3 View the contents of the data set to see how VARCHAR variables are converted to CHAR variables.

<table>
<thead>
<tr>
<th>Alphabetic List of Variables and Attributes</th>
</tr>
</thead>
<tbody>
<tr>
<td>#</td>
</tr>
<tr>
<td>---</td>
</tr>
<tr>
<td>1</td>
</tr>
<tr>
<td>2</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Alphabetic List of Variables and Attributes</th>
</tr>
</thead>
<tbody>
<tr>
<td>#</td>
</tr>
<tr>
<td>---</td>
</tr>
<tr>
<td>1</td>
</tr>
<tr>
<td>2</td>
</tr>
</tbody>
</table>

Log 2.1 Log Output

NOTE: One or more variables were converted because the data type is not supported by the V9 engine. For more details, run with options MSGLEVEL=I.
Key Ideas

- You cannot create a VARCHAR variable using an assignment statement alone. SAS sets the type to CHAR if the value on the right side of the assignment is a character string. To create a VARCHAR variable, you must define it in either a LENGTH statement or an ARRAY statement.

- The length of the CHAR variables depends on how the VARCHAR lengths are defined.

- If a VARCHAR variable is defined for its maximum length (for example, x varchar(*)), then the VARCHAR will be converted to a CHAR type with a length of 32,767 bytes.

- If a VARCHAR is defined with a specific length and then is converted to a CHAR, SAS determines the length for the CHAR based on the local SAS session encoding. SAS calculates the length of the converted VARCHAR based on the SAS session encoding. See “Avoiding Data Truncation” on page 22 and “VARCHAR Length with Implicit Type Conversion” in SAS Cloud Analytic Services: User’s Guide for more information.

See Also

- “_ALL_ ” in SAS Cloud Analytic Services: User’s Guide
- “VARCHAR Data Type” in SAS Cloud Analytic Services: User’s Guide

Group and Order a CAS Table Using the PARTITION= Data Set Option

Example Code

In this example, a SAS data set from the Sashelp library is loaded into a distributed CAS server using PROC CASUTIL. In the DATA step, the table is partitioned and then ordered within each partition using the PARTITION= and ORDERBY= data set options.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

Note: To use the ORDERBY= option, you must first specify the PARTITION= option.

```sas
proc casutil outcaslib='casuser';                  /* 1 */
   load data=sashelp.baseball replace;
run;

data mycas.baseball (partition=(team) orderby=(nHome name)); /* 2 */
   set mycas.baseball;
   keep team name nHome;
run;
proc print data=mycas.baseball;
run;
```
1. Load the Sashelp.Baseball data set to a distributed CAS table.

2. Use the PARTITION= data set option to partition the table by team. This groups the rows containing the same value for Team together, onto the same server node. The ORDERBY= option then orders the rows within each partition by nHome and Name.

**Output 2.7  Partial Output for Partitioned and Ordered Baseball Table**

<table>
<thead>
<tr>
<th>Obs</th>
<th>Name</th>
<th>Team</th>
<th>nHome</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Quinones, Rey</td>
<td>Boston</td>
<td>2</td>
</tr>
<tr>
<td>2</td>
<td>Romero, Ed</td>
<td>Boston</td>
<td>2</td>
</tr>
<tr>
<td>3</td>
<td>Barrett, Marty</td>
<td>Boston</td>
<td>4</td>
</tr>
<tr>
<td>4</td>
<td>Boggs, Wade</td>
<td>Boston</td>
<td>8</td>
</tr>
<tr>
<td>5</td>
<td>Armas, Tony</td>
<td>Boston</td>
<td>11</td>
</tr>
<tr>
<td>6</td>
<td>Gedman, Rich</td>
<td>Boston</td>
<td>16</td>
</tr>
<tr>
<td>7</td>
<td>Buckner, Bill</td>
<td>Boston</td>
<td>18</td>
</tr>
<tr>
<td>8</td>
<td>Rice, Jim</td>
<td>Boston</td>
<td>20</td>
</tr>
<tr>
<td>9</td>
<td>Evans, Dwight</td>
<td>Boston</td>
<td>26</td>
</tr>
<tr>
<td>10</td>
<td>Baylor, Don</td>
<td>Boston</td>
<td>31</td>
</tr>
<tr>
<td>11</td>
<td>Cruz, Julio</td>
<td>Chicago</td>
<td>0</td>
</tr>
<tr>
<td>12</td>
<td>Cangelosi, John</td>
<td>Chicago</td>
<td>2</td>
</tr>
<tr>
<td>13</td>
<td>Guilen, Ozzie</td>
<td>Chicago</td>
<td>2</td>
</tr>
<tr>
<td>14</td>
<td>Borilla, Bobby</td>
<td>Chicago</td>
<td>3</td>
</tr>
<tr>
<td>15</td>
<td>Tolleson, Wayne</td>
<td>Chicago</td>
<td>3</td>
</tr>
<tr>
<td>16</td>
<td>Dernier, Bob</td>
<td>Chicago</td>
<td>4</td>
</tr>
<tr>
<td>17</td>
<td>Boston, Daryl</td>
<td>Chicago</td>
<td>5</td>
</tr>
<tr>
<td>18</td>
<td>Hairston, Jerry</td>
<td>Chicago</td>
<td>5</td>
</tr>
<tr>
<td>19</td>
<td>Humphrey, Jerry</td>
<td>Chicago</td>
<td>5</td>
</tr>
<tr>
<td>20</td>
<td>Skinner, Joel</td>
<td>Chicago</td>
<td>5</td>
</tr>
</tbody>
</table>

*Note:* There is no global ordering of rows across nodes or across partitions. You should not see rows for the same team spread across different nodes or different partitions. However, you might see the global order vary. The partitioning is done first, then the ordering is done within each partition.
Key Ideas

- A partition is a group of rows that share the same values for a specified variable, or a partition key. Rows that share the same partition key are arranged together on the same worker node. After the rows are partitioned, they are ordered within each node. Rows are not ordered globally across nodes unless you run the DATA step in a single thread. See “Run the DATA Step in a Single Thread” on page 63 to see an example of running the DATA step in a single thread.

- A CAS table can be partitioned using the PARTITION= data set option and ordered within each partition using the ORDERBY= option.

- You must specify the PARTITION= option before you can specify the ORDERBY= option.

- CAS partitions tables based on the formatted values of the PARTITION= variable (rather than the raw values). For example, a table containing dates that are formatted with the QTR format is divided into 4 partitions, one partition for each yearly quarter.

- If user-defined formats are used, then the format for the specified key variable must be available to the CAS server when the table is loaded into memory.

See Also

- “PARTITION= Data Set Option” in SAS Cloud Analytic Services: User’s Guide
- “ORDERBY= Data Set Option” in SAS Cloud Analytic Services: User’s Guide
- “_ALL_ ” in SAS Cloud Analytic Services: User’s Guide

Group and Order Rows Using the DATA Step BY Statement

Example Code

This example shows how to use the BY statement in a CAS DATA step to group and order variables in a table. The BY statement groups the table rows by Make and then orders the rows within each group by Type. The example also shows how to concatenate values from multiple rows into one row. The values for car Type are concatenated into a single variable so that they can be grouped into one row. In this way, all the CarStyles for each make of car are listed together in one row, in order.

```
proc casutil; /*
  load data=sashelp.cars
  outcaslib='casuser'
  replace;
```
load the Sashelp.Cars data set to CAS.

2. Create a new variable, CarStyles, to be the aggregation variable to hold the concatenated values from other rows.

3. Retain the values for CarStyles, which are set equal to the existing variable Type. Type represents the style of car (SUV, wagon, and so on). For more information about retaining variables in the DATA step, see “RETAI N Statement” in SAS DATA Step Statements: Reference.

4. Create BY groups based on the Make of car and the style (type) of car. The values are grouped and ordered by the variables Make and Type.

5. If the DATA step encounters the first row in the BY group, then the variable CarStyles is set equal to the value for Type. The value stored in the variable CarStyles is retained from the previous row.

6. If the DATA step does not encounter the first or last row of the BY group, then the variable CarStyles equals the retained value from the previous row. Type These values are separated by concatenating a comma to the end of the first value. The TRIM function removes trailing blanks.

7. If the DATA step encounters the last row of the BY group, then execute the OUTPUT statement and write the contents of the current row to the output table.

8. Keep only the variables Make and CarStyles. The variable Type is not needed since it is now stored in the variable CarStyles.

9. Print the output.
Key Ideas

• In SAS, when using the BY statement in a DATA step, you must first sort the data. This requirement is also true for a DATA step that is running in a single thread in CAS.

• You do not have to pre-sort data to use the BY statement in a multithreaded CAS DATA step. CAS performs implicit ordering on BY variables. For example, the following DATA step is valid and pre-sorting is not required:

• On the CAS server, there is no guarantee of global ordering between BY groups. Each DATA step thread can group and order only the rows within each BY group. When a BY group is defined, the DATA step runs in multiple threads, one thread per BY group. Ordering is done only within each BY group. For more information about BY-group processing with the DATA step in CAS, see “How CAS Groups Data with BY Variables” on page 11.

See Also

• “How CAS Groups Data with BY Variables” on page 11
• “BY Statement” in SAS DATA Step Statements: Reference
• “RETAIN Statement” in SAS DATA Step Statements: Reference
• “IF-THEN/ELSE Statement” in SAS DATA Step Statements: Reference
Group Data by Two Variables

*Example Code*

This example shows how to group a table by two variables and keep only the unique rows in each group. The example first concatenates the values from two variables into a single new string variable, then orders by that new variable, keeping only the first in each group.

```sas
proc casutil;                                         /*1*/
  load data=sashelp.cars
  outcaslib='casuser'
  replace;
run; quit;

data mycas.grouped;                                   /*2*/
  set mycas.cars;
  keep make cylinders groupkey;
  groupkey = cats('make', make, 'cyl', cylinders);     /*3*/
run;

data mycas.final;                                     /*4*/
  set grouped;
  by groupkey;                                       /*5*/
  if first.groupkey then output;                     /*6*/
  drop groupkey;                                     /*7*/
run;
```

1. Load the Sashelp.Cars data set to CAS.
2. Specify the loaded CAS table as the input table and create a new table called mycas.grouped.
3. Specify the CATS function to concatenate the values in the Make and Cyl columns and store them as a concatenated string in the variable Groupkey. The CATS function also removes leading and trailing blanks.
4. Run another DATA step to isolate the second group-by variable.
5. Group the table rows by the variable Groupkey. Note: When running the DATA step in CAS, you do not have to pre-sort the data before grouping using the BY statement.
6. Specify the FIRST. automatic variable in an IF statement to identify the beginning of each BY-group. Specify the OUTPUT statement to write only the first row of each group to the output table.
7. Drop the Groupkey column.
Key Ideas

- In SAS, when using the BY statement in a DATA step, you must first sort the data. This requirement is also true for a DATA step that is running in a single thread in CAS.
- You do not have to pre-sort data to use the BY statement in a multithreaded CAS DATA step. CAS performs implicit ordering on BY variables.
- On the CAS server, there is no guarantee of global ordering between BY groups. Each DATA step thread can group and order only the rows within each BY group. When a BY group is defined, the DATA step runs in multiple threads, one thread per BY group. Ordering is done only within each BY group. For more information about BY-group processing with the DATA step in CAS, see “How CAS Groups Data with BY Variables” on page 11.

See Also

- “How CAS Groups Data with BY Variables” on page 11
- “BY Statement” in SAS DATA Step Statements: Reference
- “RETAIN Statement” in SAS DATA Step Statements: Reference
- “IF-THEN/ELSE Statement” in SAS DATA Step Statements: Reference
- “How SAS Identifies the Beginning and End of a BY Group” in SAS DATA Step Statements: Reference
- “Combining Multiple Observations and Grouping Them Based on One BY Value” in SAS DATA Step Statements: Reference
- CASLIB _ALL_ statement

Create a DATA Step Hash Object with VARCHAR Key and Data Variables

Example Code

In the following example, a hash object is created in which VARCHAR variables are defined for the KEY and DATA values. The DATA step processes in CAS. The LET macro statement substitutes the word dataopts for the SESSREF= option. When &dataopts is specified in the DATA statement, this causes the DATA step to run in CAS.

```sas
%let dataopts=%str(/sessref=casauto);
/* Simple add and iterate */
data _null_ &dataopts;
  length vc1 vc2 varchar(*);

  dcl hash h();
  h.definekey('vc1');
```
Create a DATA Step Hash Object with VARCHAR Key and Data Variables

h.definedata('vc2');
h.definedone();

vc1 = 'abcd'; vc2 = 'ABCD'; h.add();
vc1 = 'efghijk'; vc2 = 'EFGHIJK'; h.add();
vc1 = 'lmn'; vc2 = 'LMN'; h.add();

do vc1 = 'lmn', 'opqrst', 'efghijk', 'abcd';
rc = h.find();
if rc ^= 0 then
   put 'Didn''t find key ' vc1=;
else
   put _all_;
end;

put '-----------------------------';
dcl hiter hi('h');
rc = hi.next();
do until (rc ^= 0);
   put _all_;
   rc = hi.next();
end;
run;

/* Add with parameters */
data _null_ &dataopts;
   length vc1 vc2 varchar(*);
   length vc1b vc2b varchar(*);

dcl hash h();
h.definekey('vc1');
h.definedata('vc2');
h.definedone();

vc1b = 'abcd'; vc2b = 'ABCD'; h.add(key:vc1b, data:vc2b);
vc1b = 'efghijk'; vc2b = 'EFGHIJK'; h.add(key:vc1b, data:vc2b);
vc1b = 'lmn'; vc2b = 'LMN'; h.add(key:vc1b, data:vc2b);

vc1b = 'lmn', 'opqrst', 'efghijk', 'abcd';
rc = h.find(key:vc1b);
if rc ^= 0 then
   put 'Didn''t find key ' vc1b=;
else
   put _all_;
end;

put '-----------------------------';
dcl hiter hi('h');
rc = hi.next();
do until (rc ^= 0);
   put _all_;
   rc = hi.next();
end;
run;
Key Ideas

- You can use VARCHAR variables for KEY and DATA values when defining a hash object for a DATA step running in CAS or in SAS. VARCHAR variables are supported with all hash object methods where character values are supported.

- Ordered hashes are not supported in CAS (for example, `dcl hash h(ordered: 'yes');`) is not supported in the DATA step running in CAS.

- A fixed-width, 32-byte SHA-256 hash digest value is computed for VARCHAR keys.

See Also

- SAS Component Objects: Reference
- “_ALL_” in SAS Cloud Analytic Services: User’s Guide
Run the DATA Step in a Single Thread

Example Code

The following example shows how to use the SINGLE= option to run the DATA step in a single thread. This option is useful for running functions that have inter-row dependencies.

```sas
data mycas.shoes / single=yes;                  /* */
    set mycas.shoes;
    put "The number of Threads is " _nthreads_;  /* */
run;
```

1 Using the Mycas.Shoes data set from the previous example, load the Mycas.Shoes table. Specify the SINGLE=YES option.

2 Specify the automatic variable, _NTHREADS_ with the PUT statement to verify the number of threads that are processing.

Output 2.10 Partial Log Output

```
NOTE: Running DATA step in Cloud Analytic Services.
The number of Threads is 1
```

Key Ideas

- The DATA step in CAS automatically runs in multiple threads on every available computer in your system by default. To run the DATA step in multiple threads, specify the SINGLE=YES option.
- A SAS DATA step (a DATA step that is not executing in CAS) always runs in a single thread.
- For more information about CAS sessions, see “Sessions” in SAS Cloud Analytic Services: Fundamentals.

See Also

- SESSOPTS=
- SINGLE= in SAS DATA Step Statements: Reference
- “Inter-row Dependencies in DATA Step Processing” on page 11
- “DATA Statement” in SAS DATA Step Statements: Reference
- “_ALL_” in SAS Cloud Analytic Services: User’s Guide
View DATA Step Processing Information Using Automatic Variables

Example Code

This first example uses the _THREADID_ automatic variable to get information about a multithreaded DATA step that contains two BY variables. When the DATA step runs in CAS, the table is grouped and distributed by the first BY variable, _make_. Each BY group is then ordered by the second BY variable, _type_. The FIRST. automatic variable is used to show the distinct BY groups in the output table.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
data mycas.cars;
  set sashelp.cars;
run;

data mycas.cars;
  set mycas.cars(where=(weight>4000));
  by make type;
  if first.make then first="BY Group";
  keep make type first threadid;
  threadid = _threadid_
run;
proc print data=mycas.cars;
  title 'Cars By Make By Type - Multithreaded'; run; title;
run;
```

<table>
<thead>
<tr>
<th>Obs</th>
<th>Make</th>
<th>Type</th>
<th>First</th>
<th>Threadid</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>BMW</td>
<td>Sedan</td>
<td>BY Group</td>
<td>2</td>
</tr>
<tr>
<td>2</td>
<td>BMW</td>
<td>Sedan</td>
<td></td>
<td>2</td>
</tr>
<tr>
<td>3</td>
<td>BMW</td>
<td>SUV</td>
<td></td>
<td>2</td>
</tr>
<tr>
<td>4</td>
<td>BMW</td>
<td>SUV</td>
<td></td>
<td>2</td>
</tr>
<tr>
<td>5</td>
<td>Honda</td>
<td>SUV</td>
<td>BY Group</td>
<td>3</td>
</tr>
<tr>
<td>6</td>
<td>Jaguar</td>
<td>Sedan</td>
<td>BY Group</td>
<td>11</td>
</tr>
<tr>
<td>7</td>
<td>Jaguar</td>
<td>Sports</td>
<td></td>
<td>11</td>
</tr>
<tr>
<td>8</td>
<td>Cadillac</td>
<td>Sedan</td>
<td>BY Group</td>
<td>20</td>
</tr>
<tr>
<td>9</td>
<td>Cadillac</td>
<td>Sedan</td>
<td></td>
<td>20</td>
</tr>
<tr>
<td>10</td>
<td>Cadillac</td>
<td>SUV</td>
<td></td>
<td>20</td>
</tr>
<tr>
<td>11</td>
<td>Cadillac</td>
<td>Truck</td>
<td></td>
<td>20</td>
</tr>
<tr>
<td>12</td>
<td>Dodge</td>
<td>Sedan</td>
<td>BY Group</td>
<td>23</td>
</tr>
<tr>
<td>13</td>
<td>Dodge</td>
<td>Sedan</td>
<td></td>
<td>23</td>
</tr>
<tr>
<td>14</td>
<td>Dodge</td>
<td>Truck</td>
<td></td>
<td>23</td>
</tr>
<tr>
<td>15</td>
<td>Infiniti</td>
<td>Wagon</td>
<td>BY Group</td>
<td>23</td>
</tr>
<tr>
<td>16</td>
<td>Infiniti</td>
<td>Wagon</td>
<td></td>
<td>23</td>
</tr>
<tr>
<td>17</td>
<td>Buick</td>
<td>SUV</td>
<td>BY Group</td>
<td>29</td>
</tr>
<tr>
<td>18</td>
<td>Buick</td>
<td>SUV</td>
<td></td>
<td>29</td>
</tr>
<tr>
<td>19</td>
<td>Volkswagen</td>
<td>Sedan</td>
<td>BY Group</td>
<td>32</td>
</tr>
<tr>
<td>20</td>
<td>Volkswagen</td>
<td>Sedan</td>
<td></td>
<td>32</td>
</tr>
<tr>
<td>21</td>
<td>Volkswagen</td>
<td>SUV</td>
<td></td>
<td>32</td>
</tr>
<tr>
<td>22</td>
<td>Volkswagen</td>
<td>Wagon</td>
<td></td>
<td>32</td>
</tr>
<tr>
<td>23</td>
<td>Pontiac</td>
<td>Sedan</td>
<td>BY Group</td>
<td>43</td>
</tr>
<tr>
<td>24</td>
<td>Land Rover</td>
<td>SUV</td>
<td>BY Group</td>
<td>44</td>
</tr>
<tr>
<td>25</td>
<td>Land Rover</td>
<td>SUV</td>
<td></td>
<td>44</td>
</tr>
</tbody>
</table>
You can also get information about the node (host) name and the number of threads the DATA step is running in by specifying the _HOSTNAME_ and _NTHREADS_ automatic variables. In the following example, the host name and thread count results are printed in the SAS log.

```sas
data mycas.cars;
set mycas.cars;
put _hostname_
put _nthreads_
run;
```

**Output 2.11**  Log Output for View DATA Step Processing Information Using Automatic Variables

```sas
1 OPTIONS NONOTES NOSTIMER NOSOURCE NOSYNTAXCHECK;
72
73 data mycas.cars;
74 set mycas.cars;
75 put _hostname_
76 put _nthreads_
77 run;
```

NOTE: Running DATA step in Cloud Analytic Services.
NOTE: The DATA step will run in multiple threads.
srvrnode002
srvrnode005
srvrnode007
192
srvrnode003
srvrnode004
srvrnode006
NOTE: Duplicate messages output by DATA step:
srvrnode002  (occurred 22 times)
srvrnode005  (occurred 27 times)
srvrnode007  (occurred 12 times)
192  (occurred 103 times)
srvrnode003  (occurred 17 times)
srvrnode004  (occurred 14 times)
srvrnode006  (occurred 11 times)
NOTE: There were 103 observations read from the table CARS in caslib CASUSER(sasdemo).
NOTE: The table cars in caslib CASUSER(sasdemo) has 103 observations and 4 variables.
NOTE: DATA statement used (Total process time):
real time 0.42 seconds
cpu time 0.03 seconds
Key Ideas

- A DATA step that has an input table automatically runs in multiple threads in CAS. This is the case even in a single-machine CAS environment. By default, the DATA step runs in all available threads on every computer node in the cluster.

- Single-threaded DATA step processing is useful on small- to medium-sized tables where performance is not an issue and where preserving order or maintaining inter-row dependencies is important.

- When the DATA step runs in CAS, the same DATA step program is replicated across the CAS cluster. On each node in the cluster, the DATA step runs multiple threads. Each thread runs on only a portion of the table.

- There are restrictions on what language elements you can use in a DATA step running in CAS. For information about these restrictions, see “Restrictions for Language Elements in the DATA Step” on page 15.

See Also

- “Automatic DATA Step Variables” on page 19
- “DATA Statement” in SAS DATA Step Statements: Reference
- “Controlling Where the DATA Step Runs” on page 5
- “About SAS Cloud Analytic Services Sessions” in SAS Cloud Analytic Services: Fundamentals
- “_ALL_” in SAS Cloud Analytic Services: User’s Guide

Use the Macro Facility to Generate CAS DATA Step Code

Example Code

The following example uses a SAS macro to generate SAS code. The macro takes the tables listed in the TABLES= parameter and writes them out as a SAS data set. Macros compile on the SAS server.

For information about how to get started with the examples in this document, see “Set Up Code for Examples” on page 2.

```sas
data mycas.class; /* 1 */
   set sashelp.class; run;
data mycas.cars;
   set sashelp.cars; run;
data mycas.air;
   set sashelp.air; run;
libname mysas */pathname/mucas/"; /* 2 */
```
%macro load(tables=);                                              /* 3 */
%let n=%sysfunc(countw(&tables,%str( )));                          /* 4 */
%do i=1 %to &n;                                                    /* 5 */
    %let dsn=%scan(&tables,&i);
    data mysas.&dsn;                                                /* 6 */
    set mycas.&dsn;
    run;
%end;
%mend;                                                             /* 7 */
%load(tables=class cars air);                                        

1  Use the DATA step to load three SASHELP data sets to CAS.
2  Create a SAS libref.
3  Start the macro definition.
4  Set \n\ equal to the number of tables.
5  Loop through \n\ times (for each table).
6  Create a SAS data set from each CAS table.
7  End macro definition.

Log 2.2  Partial Log Output for Macro

NOTE: Libref MYSAS was successfully assigned as follows:
   Engine:        V9
   Physical Name: /pathname/mycas

NOTE: There were 19 observations read from the data set MYCAS.CLASS.
NOTE: The data set MYSAS.CLASS has 19 observations and 5 variables.
NOTE: DATA statement used (Total process time):
   real time          0.02 seconds
   cpu time           0.01 seconds

NOTE: There were 428 observations read from the data set MYCAS.CARS.
NOTE: The data set MYSAS.CARS has 428 observations and 15 variables.
NOTE: DATA statement used (Total process time):
   real time          0.02 seconds
   cpu time           0.00 seconds

NOTE: There were 144 observations read from the data set MYCAS.AIR.
NOTE: The data set MYSAS.AIR has 144 observations and 2 variables.
NOTE: DATA statement used (Total process time):
   real time          0.02 seconds
   cpu time           0.01 seconds

See Also

• “Using the Macro Facility in SAS Viya” in SAS Macro Language: Reference
• “CAS Statement” in SAS Cloud Analytic Services: User’s Guide
• “_ALL_” in SAS Cloud Analytic Services: User’s Guide
DATA Step Program Walk-through

Example Code

The following example shows the steps that you need to perform to run the DATA step in CAS.

1. **Start a CAS session.**

   ```sas
   386  caslib _all_ assign;
   NOTE: A SAS Library associated with a caslib can only reference library member
   names that conform
   to SAS Library naming conventions.
   NOTE: CASLIB CASUSER(sasuser) for session CASAUTO will be mapped to SAS Library
   CASUSER.
   NOTE: CASLIB CASUSERHDFS(sasuser) for session CASAUTO will be mapped to SAS
   Library CASUSERH.
   387  libname mycas cas;
   NOTE: Libref MYCAS was successfully assigned as follows:
   Engine:        CAS
   ```

   The SESSOPTS= option in the CAS statement is used with the CASLIB= session option to ensure that the CASUSER personal caslib is set as the active caslib.

   The `CASLIB_ALL.Assign` maps caslibs to SAS libraries. Caslibs provide a way to access in-memory tables and an associated data source. In this example, the personal caslib CASUSER is being used, so no CASLIB statement is needed. For more information about the CASUSER and CASUSERHDFS personal caslibs, see Personal, Predefined, and Custom Caslibs in SAS Cloud Analytic Services: Fundamentals: “Caslibs, Files, and Tables” in *SAS Cloud Analytic Services: Fundamentals*.

   The LIBNAME CAS statement creates a SAS library that is associated with the CAS engine. Mycas is the CAS engine libref.

2. **Load data to CAS.**

   Use the `CASUTIL` procedure to load data to CAS. The data is loaded from the Sashelp library in the local SAS client session into an in-memory table in CAS. The DATA step processes in SAS. The output table, Cars, is loaded into the CAS server as an in-memory table.

   ```sas
   proc casutil;
   load data=sashelp.cars
   outcaslib='casuser'
   replace;
   run; quit;
   ```

   Here is the log output:
3. **Run the DATA step in CAS.** The DATA step runs in CAS in multiple threads by default.

```sas
data mycas.cars2(promote=yes)/sessref=casauto;         /* 1 */
set mycas.cars;                                     /* 2 */
  format Combined 2.;
  Combined = (MPG_Highway*.45) + (MPG_City*.55);   /* 3 */
  if type="SUV" and origin="USA";                  /* 4 */
  keep make type MPG_City MPG_Highway Combined;
by make;
run;
title 'Combined Average MPG for SUVs in the USA By Make';
proc print data=mycas.cars2; run;                      /* 5 */
```

1. Run the DATA step in CAS and specify Cars2 as the output table. Promote the output table to global scope by specifying the `PROMOTE=` data set option. Global scope lets other sessions access the table.

2. Specify the in-memory table that you loaded in 2. Load data to CAS, on page 68 as the input table.

3. Create a new variable, Combined, to calculate the combined average miles per gallon for SUVs in the USA.

4. Specify the **KEEP statement** to keep some variables in the output table and drop the rest. Use a subsetting **IF statement** to subset the data and select only rows containing SUV for Type and USA for Origin.

5. Print the results.

Here is the PROC PRINT output.
### Combined Average MPG for SUVs in the USA By Make

<table>
<thead>
<tr>
<th>Obs</th>
<th>Make</th>
<th>Type</th>
<th>MPG_City</th>
<th>MPG_Highway</th>
<th>Combined</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Cadillac</td>
<td>SUV</td>
<td>16</td>
<td>21</td>
<td>18</td>
</tr>
<tr>
<td>2</td>
<td>Cadillac</td>
<td>SUV</td>
<td>14</td>
<td>18</td>
<td>16</td>
</tr>
<tr>
<td>3</td>
<td>Saturn</td>
<td>SUV</td>
<td>21</td>
<td>26</td>
<td>23</td>
</tr>
<tr>
<td>4</td>
<td>Dodge</td>
<td>SUV</td>
<td>15</td>
<td>21</td>
<td>18</td>
</tr>
<tr>
<td>5</td>
<td>Buick</td>
<td>SUV</td>
<td>19</td>
<td>26</td>
<td>22</td>
</tr>
<tr>
<td>6</td>
<td>Buick</td>
<td>SUV</td>
<td>15</td>
<td>21</td>
<td>18</td>
</tr>
<tr>
<td>7</td>
<td>Pontiac</td>
<td>SUV</td>
<td>19</td>
<td>26</td>
<td>22</td>
</tr>
<tr>
<td>8</td>
<td>Mercury</td>
<td>SUV</td>
<td>16</td>
<td>21</td>
<td>18</td>
</tr>
<tr>
<td>9</td>
<td>Lincoln</td>
<td>SUV</td>
<td>13</td>
<td>18</td>
<td>15</td>
</tr>
<tr>
<td>10</td>
<td>Lincoln</td>
<td>SUV</td>
<td>13</td>
<td>18</td>
<td>15</td>
</tr>
<tr>
<td>11</td>
<td>Chevrolet</td>
<td>SUV</td>
<td>19</td>
<td>22</td>
<td>20</td>
</tr>
<tr>
<td>12</td>
<td>Chevrolet</td>
<td>SUV</td>
<td>14</td>
<td>18</td>
<td>16</td>
</tr>
<tr>
<td>13</td>
<td>Chevrolet</td>
<td>SUV</td>
<td>16</td>
<td>21</td>
<td>18</td>
</tr>
<tr>
<td>14</td>
<td>Chevrolet</td>
<td>SUV</td>
<td>14</td>
<td>18</td>
<td>16</td>
</tr>
<tr>
<td>15</td>
<td>GMC</td>
<td>SUV</td>
<td>16</td>
<td>19</td>
<td>17</td>
</tr>
<tr>
<td>16</td>
<td>GMC</td>
<td>SUV</td>
<td>13</td>
<td>17</td>
<td>15</td>
</tr>
<tr>
<td>17</td>
<td>GMC</td>
<td>SUV</td>
<td>15</td>
<td>19</td>
<td>17</td>
</tr>
<tr>
<td>18</td>
<td>Jeep</td>
<td>SUV</td>
<td>20</td>
<td>24</td>
<td>22</td>
</tr>
<tr>
<td>19</td>
<td>Jeep</td>
<td>SUV</td>
<td>16</td>
<td>19</td>
<td>17</td>
</tr>
<tr>
<td>20</td>
<td>Jeep</td>
<td>SUV</td>
<td>16</td>
<td>21</td>
<td>18</td>
</tr>
<tr>
<td>21</td>
<td>Hummer</td>
<td>SUV</td>
<td>10</td>
<td>12</td>
<td>11</td>
</tr>
<tr>
<td>22</td>
<td>Ford</td>
<td>SUV</td>
<td>16</td>
<td>23</td>
<td>20</td>
</tr>
<tr>
<td>23</td>
<td>Ford</td>
<td>SUV</td>
<td>10</td>
<td>13</td>
<td>11</td>
</tr>
<tr>
<td>24</td>
<td>Ford</td>
<td>SUV</td>
<td>15</td>
<td>20</td>
<td>17</td>
</tr>
<tr>
<td>25</td>
<td>Ford</td>
<td>SUV</td>
<td>15</td>
<td>19</td>
<td>17</td>
</tr>
</tbody>
</table>

Here is the log output:
348  data mycas.cars2(promote=yes)/sessref=casauto;
349  set mycas.cars;
350    format Combined 2.2;
351    Combined = (MPG_Highway*.45) + (MPG_City*.55);
352    if type="SUV" and origin="USA";
353    keep make type MPG_City MPG_Highway Combined;
354    by make;
355  run;
356
NOTE: Running DATA step in Cloud Analytic Services.
NOTE: There were 428 observations read from the table CARS in caslib CASUSER(sasuser).
NOTE: The table cars2 in caslib CASUSER(sasuser) has 25 observations and 5 variables.
NOTE: DATA statement used (Total process time):
real time          0.34 seconds
cpu time           0.03 seconds

357  title 'Combined Average MPG for SUVs in the USA By Make';
358  proc print data=mycas.cars2; run;
359
NOTE: PROCEDURE PRINT used (Total process time):
real time          0.11 seconds
cpu time           0.04 seconds

4. Save the table and list information about the table.

The in-memory CAS table, Cars2, can be saved permanently to disk in the server. Use the SAVE statement in the CASUTIL procedure to save the table. Use the CONTENTS statement to view information about the table. The REPLACE statement enables you to run the same procedure again and overwrite the table without an error.

```
proc casutil;
  save casdata='cars2'
    outcaslib='casuser' replace;
  contents casdata='cars2';
  run; quit;
```

Here is the log output:

<table>
<thead>
<tr>
<th>Table Name</th>
<th>Number of Rows</th>
<th>Number of Columns</th>
<th>RLS Version</th>
<th>Created</th>
<th>Last Modified</th>
<th>Protected Table</th>
<th>Repeated Table</th>
<th>View</th>
<th>Source Name</th>
<th>Compressed</th>
</tr>
</thead>
<tbody>
<tr>
<td>CARS2</td>
<td>25</td>
<td>5</td>
<td>2017-01-30</td>
<td>Yes</td>
<td>No</td>
<td>No</td>
<td>No</td>
<td>_HBRVNYD9VWI_EIDSFH2W_2Y20XK9XN1</td>
<td>No</td>
<td></td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Node</th>
<th>Number of Blocks</th>
<th>Active Blocks</th>
<th>Rows</th>
<th>Fixed Data size</th>
<th>Variable Data size</th>
<th>Blocks Mapped</th>
<th>Memory Mapped</th>
<th>Blocks Unmapped</th>
<th>Memory Unmapped</th>
<th>Blocks Allocated</th>
<th>Memory Allocated</th>
</tr>
</thead>
<tbody>
<tr>
<td>ALL</td>
<td>20</td>
<td>15</td>
<td>25</td>
<td>1200</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td></td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Column</th>
<th>Label</th>
<th>Type</th>
<th>Length</th>
<th>Format Name</th>
<th>Formatted Length</th>
<th>Format Width</th>
<th>Format Decimal</th>
</tr>
</thead>
</table>
5. **Drop the promoted table from memory.** You cannot replace a promoted table with the same name, so you must drop the table first if you want to replace it.

   ```
   proc casutil;
   droptable casdata='cars2';
   run; quit;
   ```

   **Note:** You can also delete tables in SAS Studio by using the Libraries tab in the left pane. Right-click on the table that you want to delete, and select delete from the popup menu.

   Here is the entire program together.

   ```
   options cashost="srvrnode001" casport=10380;
   cas casauto sessopts=(caslib='casuser');
   libname mycas cas;
   caslib _all_ assign;
   proc casutil;
   load data=sashelp.cars
   outcaslib='casuser'
   replace;
   run; quit;
   ```

   ```
   data mycas.cars2(promote=yes)/sessref=casauto;
   set mycas.cars;
   format Combined 2.;
   Combined = (MPG_Highway*.45) + (MPG_City*.55);
   if type="SUV" and origin="USA";
   keep make type MPG_City MPG_Highway Combined;
   ```
by make;
run;

title 'Combined Average MPG for SUVs in the USA By Make';
proc print data=mycas.cars2; run;

proc casutil;
    save casdata='cars2'
    outcaslib='casuser' replace;
    contents casdata='cars2';
run; quit;

proc casutil;
    droptable casdata='cars2';
run; quit;

**Key Ideas**

- You can use many of the same statements that you use in a traditional SAS DATA step in CAS. For example, the SET, KEEP, and DROP statements perform the same functions in CAS as they do in a SAS DATA step. For a complete list of supported language elements, see “SAS Language Elements Support for the CAS DATA Step” on page 14.

- When the DATA step runs in CAS, it runs where the tables are stored. Running a DATA step in CAS means that the DATA step code and all the statements within it are executing in a CAS server session. For information about how the DATA step runs in CAS, see “The DATA Step and CAS” on page 3.

- When the DATA step runs in CAS, it runs in all available threads by default. See “Controlling Where the DATA Step Runs” on page 5 for more information.

- For a list of requirements for running the DATA step in CAS, see “Requirements for Running the DATA Step in CAS” on page 4.

- You can use the CASUTIL procedure to do more than load your SAS data sets into CAS. For more information, see “CASUTIL Procedure” in *SAS Cloud Analytic Services: User’s Guide*.

**See Also**

- “CAS Statement” in *SAS Cloud Analytic Services: User’s Guide*
- “_ALL_” in *SAS Cloud Analytic Services: User’s Guide*
- “Accessing Data” in *SAS Cloud Analytic Services: User’s Guide*
- “CASUTIL Procedure” in *SAS Cloud Analytic Services: User’s Guide*
- *An Introduction to SAS Viya Programming*
Gain Greater Insight into Your SAS® Software with SAS Books.

Discover all that you need on your journey to knowledge and empowerment.